Hyperbolic plane $\mathbb{H}^2$

This Jupyter notebook illustrates some differential geometry capabilities of SageMath on the example of the hyperbolic plane. The corresponding tools have been developed within the SageManifolds (https://sagemanifolds.obspm.fr) project.

A version of SageMath at least equal to 7.5 is required to run this notebook:

In [1]:
```
version()
```
Out[1]:
```
'SageMath version 9.2, Release Date: 2020-10-24'
```

First we set up the notebook to display mathematical objects using LaTeX formatting:

In [2]:
```
%display latex
```

We also tell Maxima, which is used by SageMath for simplifications of symbolic expressions, that all computations involve real variables:

In [3]:
```
maxima_calculus.eval("domain: real;")
```
Out[3]:
```
real
```

We declare $\mathbb{H}^2$ as a 2-dimensional differentiable manifold:

In [4]:
```
H2 = Manifold(2, 'H2', latex_name=r'\mathbb{H}^2', start_index=1)
print(H2)
H2
```
```
2-dimensional differentiable manifold H2
```
Out[4]: $\mathbb{H}^2$

We shall introduce charts on $\mathbb{H}^2$ that are related to various models of the hyperbolic plane as submanifolds of $\mathbb{R}^3$. Therefore, we start by declaring $\mathbb{R}^3$ as a 3-dimensional manifold equiped with a global chart: the chart of Cartesian coordinates (X, Y, Z):

In [5]:
```
R3 = Manifold(3, 'R3', latex_name=r'\mathbb{R}^3', start_index=1)
X3.<X,Y,Z> = R3.chart()
X3
```
Out[5]: $\left(\mathbb{R}^3,(X, Y, Z)\right)$

Hyperboloid model

The first chart we introduce is related to the **hyperboloid model of** $\mathbb{H}^2$, namely to the representation of $\mathbb{H}^2$ as the upper sheet ($Z > 0$) of the hyperboloid of two sheets defined in $\mathbb{R}^3$ by the equation $X^2 + Y^2 - Z^2 = -1$:

In [6]:
```
X_hyp.<X,Y> = H2.chart()
X_hyp
```
Out[6]: $\left(\mathbb{H}^2,(X, Y)\right)$

The corresponding embedding of $\mathbb{H}^2$ in $\mathbb{R}^3$ is

In [7]:
```
Phi1 = H2.diff_map(R3, [X, Y, sqrt(1+X^2+Y^2)], name='Phi_1', latex_name=r'\Phi_1')
Phi1.display()
```

Out[7]:

$$\begin{array}{llcl} \Phi_1: & \mathbb{H}^2 & \longrightarrow & \mathbb{R}^3 \\ & (X,Y) & \longmapsto & (X,Y,Z) = \left(X, Y, \sqrt{X^2+Y^2+1}\right) \end{array}$$

By plotting the chart $\left(\mathbb{H}^2, (X,Y)\right)$ in terms of the Cartesian coordinates of $\mathbb{R}^3$, we get a graphical view of $\Phi_1(\mathbb{H}^2)$:

In [8]:
```
show(X_hyp.plot(X3, mapping=Phi1, number_values=15, color='blue'),
     aspect_ratio=1, figsize=7)
```



A second chart is obtained from the polar coordinates (r, φ) associated with (X, Y). Contrary to (X, Y), the polar chart is not defined on the whole $\mathbb{H}^2$, but on the complement U of the segment $\{Y = 0, x \geq 0\}$:

In [9]:
```
U = H2.open_subset('U', coord_def={X_hyp: (Y!=0, X<0)})
print(U)
```

```
Open subset U of the 2-dimensional differentiable manifold H2
```

Note that (y!=0, x<0) stands for $y \neq 0$ OR $x < 0$; the condition $y \neq 0$ AND $x < 0$ would have been written [y!=0, x<0] instead.

In [10]:
```
X_pol.<r,ph> = U.chart(r'r:(0,+oo) ph:(0,2*pi):\varphi')
X_pol
```

Out[10]: $(U, (r, \varphi))$

In [11]:
```
X_pol.coord_range()
```

Out[11]: $r: \ (0, +\infty); \quad \varphi: \ (0, 2\pi)$

We specify the transition map between the charts $(U, (r, \varphi))$ and $\left(\mathbb{H}^2, (X,Y)\right)$ as $X = r\cos\varphi$, $Y = r\sin\varphi$:

In [12]:
```
pol_to_hyp = X_pol.transition_map(X_hyp, [r*cos(ph), r*sin(ph)])
pol_to_hyp
```

Out[12]: $(U, (r, \varphi)) \rightarrow (U, (X, Y))$

In [13]:
```
pol_to_hyp.display()
```

Out[13]:

$$\left\{\begin{array}{lcl} X & = & r\cos(\varphi) \\ Y & = & r\sin(\varphi) \end{array}\right.$$

In [14]:
```
pol_to_hyp.set_inverse(sqrt(X^2+Y^2), atan2(Y, X))
```

```
Check of the inverse coordinate transformation:
  r == r  *passed*
  ph == arctan2(r*sin(ph), r*cos(ph))  **failed**
  X == X  *passed*
  Y == Y  *passed*
NB: a failed report can reflect a mere lack of simplification.
```

In [15]:
```
pol_to_hyp.inverse().display()
```

Out[15]:

$$\left\{\begin{array}{lcl} r & = & \sqrt{X^2+Y^2} \\ \varphi & = & \arctan(Y, X) \end{array}\right.$$

The restriction of the embedding Φ_1 to U has then two coordinate expressions:

In [16]:
```
Phi1.restrict(U).display()
```

Out[16]:

$$\begin{array}{llcl} \Phi_1: & U & \longrightarrow & \mathbb{R}^3 \\ & (X,Y) & \longmapsto & (X,Y,Z) = \left(X, Y, \sqrt{X^2+Y^2+1}\right) \\ & (r,\varphi) & \longmapsto & (X,Y,Z) = \left(r\cos(\varphi), r\sin(\varphi), \sqrt{r^2+1}\right) \end{array}$$

In [17]:
```
graph_hyp = X_pol.plot(X3, mapping=Phi1.restrict(U), number_values=15, ranges={r: (0, 3)},
                       color='blue')
show(graph_hyp, aspect_ratio=1, figsize=7)
```



In [18]:
```
Phi1._coord_expression
```

Out[18]:

$$\left\{\left(\left(\mathbb{H}^2,(X,Y)\right),\left(\mathbb{R}^3,(X,Y,Z)\right)\right): \left(X, Y, \sqrt{X^2+Y^2+1}\right)\right\}$$

Metric and curvature

The metric on $\mathbb{H}^2$ is that induced by the Minkowksy metric on $\mathbb{R}^3$:

$$\eta = \mathrm{d}X \otimes \mathrm{d}X + \mathrm{d}Y \otimes \mathrm{d}Y - \mathrm{d}Z \otimes \mathrm{d}Z$$

In [19]:
```
eta = R3.lorentzian_metric('eta', latex_name=r'\eta')
eta[1,1] = 1 ; eta[2,2] = 1 ; eta[3,3] = -1
eta.display()
```

Out[19]: $\eta = \mathrm{d}X \otimes \mathrm{d}X + \mathrm{d}Y \otimes \mathrm{d}Y - \mathrm{d}Z \otimes \mathrm{d}Z$

In [20]:
```
g = H2.metric('g')
g.set( Phi1.pullback(eta) )
g.display()
```

Out[20]:
$$g = \left(\frac{Y^2+1}{X^2+Y^2+1} \right) \mathrm{d}X \otimes \mathrm{d}X + \left(-\frac{XY}{X^2+Y^2+1} \right) \mathrm{d}X \otimes \mathrm{d}Y + \left(-\frac{XY}{X^2+Y^2+1} \right) \mathrm{d}Y \otimes \mathrm{d}X + \left(\frac{}{X} \right.$$

The expression of the metric tensor in terms of the polar coordinates is

In [21]:
```
g.display(X_pol.frame(), X_pol)
```

Out[21]:
$$g = \left(\frac{1}{r^2+1} \right) \mathrm{d}r \otimes \mathrm{d}r + r^2 \mathrm{d}\varphi \otimes \mathrm{d}\varphi$$

The Riemann curvature tensor associated with g is

In [22]:
```
Riem = g.riemann()
print(Riem)
```

```
Tensor field Riem(g) of type (1,3) on the 2-dimensional differentiable manifold H2
```

In [23]:
```
Riem.display(X_pol.frame(), X_pol)
```

Out[23]:
$$\mathrm{Riem}\,(g) = -r^2 \frac{\partial}{\partial r} \otimes \mathrm{d}\varphi \otimes \mathrm{d}r \otimes \mathrm{d}\varphi + r^2 \frac{\partial}{\partial r} \otimes \mathrm{d}\varphi \otimes \mathrm{d}\varphi \otimes \mathrm{d}r + \left(\frac{1}{r^2+1} \right) \frac{\partial}{\partial \varphi} \otimes \mathrm{d}r \otimes \mathrm{d}r \otimes \mathrm{d}\varphi + \left(- \right.$$

The Ricci tensor and the Ricci scalar:

In [24]:
```
Ric = g.ricci()
print(Ric)
```

```
Field of symmetric bilinear forms Ric(g) on the 2-dimensional differentiable manifold
H2
```

In [25]:
```
Ric.display(X_pol.frame(), X_pol)
```

Out[25]:
$$\mathrm{Ric}\,(g) = \left(-\frac{1}{r^2+1} \right) \mathrm{d}r \otimes \mathrm{d}r - r^2 \mathrm{d}\varphi \otimes \mathrm{d}\varphi$$

In [26]:
```
Rscal = g.ricci_scalar()
print(Rscal)
```

```
Scalar field r(g) on the 2-dimensional differentiable manifold H2
```

In [27]:
```
Rscal.display()
```

Out[27]:
$$\begin{array}{llcl} \mathrm{r}\,(g) : & \mathbb{H}^2 & \longrightarrow & \mathbb{R} \\ & (X,Y) & \longmapsto & -2 \\ \text{on } U : & (r,\varphi) & \longmapsto & -2 \end{array}$$

Hence we recover the fact that $(\mathbb{H}^2, g)$ is a space of **constant negative curvature**.

In dimension 2, the Riemann curvature tensor is entirely determined by the Ricci scalar R according to

$$R^i{}_{jlk} = \frac{R}{2}\left(\delta^i{}_k g_{jl} - \delta^i{}_l g_{jk}\right)$$

Let us check this formula here, under the form $R^i{}_{jlk} = -R g_{j[k}\delta^i{}_{l]}$:

In [28]:
```
delta = H2.tangent_identity_field()
Riem == - Rscal*(g*delta).antisymmetrize(2,3)  # 2,3 = last positions of the type-(1,
3) tensor g*delta
```

Out[28]: True

Similarly the relation $\mathrm{Ric} = (R/2)\, g$ must hold:

In [29]:
```
Ric == (Rscal/2)*g
```

Out[29]: True

Poincaré disk model

The Poincaré disk model of $\mathbb{H}^2$ is obtained by stereographic projection from the point $S = (0, 0, -1)$ of the hyperboloid model to the plane $Z = 0$. The radial coordinate R of the image of a point of polar coordinate (r, φ) is

$$R = \frac{r}{1 + \sqrt{1 + r^2}}.$$

Hence we define the Poincaré disk chart on $\mathbb{H}^2$ by

In [30]:
```
X_Pdisk.<R,ph> = U.chart(r'R:(0,1) ph:(0,2*pi):\varphi')
X_Pdisk
```

Out[30]: $(U, (R, \varphi))$

In [31]:
```
X_Pdisk.coord_range()
```

Out[31]: $R :\ (0, 1) ; \quad \varphi :\ (0, 2\pi)$

and relate it to the hyperboloid polar chart by

In [32]:
```
pol_to_Pdisk = X_pol.transition_map(X_Pdisk, [r/(1+sqrt(1+r^2)), ph])
pol_to_Pdisk
```

Out[32]: $(U, (r, \varphi)) \to (U, (R, \varphi))$

In [33]:
```
pol_to_Pdisk.display()
```

Out[33]:
$$\left\{\begin{array}{lcl} R & = & \frac{r}{\sqrt{r^2+1}+1} \\ \varphi & = & \varphi \end{array}\right.$$

In [34]:
```
pol_to_Pdisk.set_inverse(2*R/(1-R^2), ph)
pol_to_Pdisk.inverse().display()
```

Out[34]:
$$\left\{\begin{array}{lcl} r & = & -\frac{2R}{R^2-1} \\ \varphi & = & \varphi \end{array}\right.$$

A view of the Poincaré disk chart via the embedding Φ_1:

In [35]:
```
show(X_Pdisk.plot(X3, mapping=Phi1.restrict(U), ranges={R: (0,0.9)}, color='blue',
                  number_values=15),
     aspect_ratio=1, figsize=7)
```



The expression of the metric tensor in terms of coordinates (R, φ):

In [36]:
```
g.display(X_Pdisk.frame(), X_Pdisk)
```

Out[36]:
$$g = \left(\frac{4}{R^4 - 2R^2 + 1}\right) \mathrm{d}R \otimes \mathrm{d}R + \left(\frac{4R^2}{R^4 - 2R^2 + 1}\right) \mathrm{d}\varphi \otimes \mathrm{d}\varphi$$

We may factorize each metric component:

In [37]:
```
for i in [1,2]:
    g[X_Pdisk.frame(), i, i, X_Pdisk].factor()
g.display(X_Pdisk.frame(), X_Pdisk)
```

Out[37]:
$$g = \frac{4}{(R+1)^2(R-1)^2} \mathrm{d}R \otimes \mathrm{d}R + \frac{4R^2}{(R+1)^2(R-1)^2} \mathrm{d}\varphi \otimes \mathrm{d}\varphi$$

Cartesian coordinates on the Poincaré disk

Let us introduce Cartesian coordinates (u, v) on the Poincaré disk; since the latter has a unit radius, this amounts to define the following chart on $\mathbb{H}^2$:

In [38]:
```
X_Pdisk_cart.<u,v> = H2.chart('u:(-1,1) v:(-1,1)')
X_Pdisk_cart.add_restrictions(u^2+v^2 < 1)
X_Pdisk_cart
```

Out[38]: $\left(\mathbb{H}^2, (u, v)\right)$

On U, the Cartesian coordinates (u, v) are related to the polar coordinates (R, φ) by the standard formulas:

In [39]:
```
Pdisk_to_Pdisk_cart = X_Pdisk.transition_map(X_Pdisk_cart, [R*cos(ph), R*sin(ph)])
Pdisk_to_Pdisk_cart
```

Out[39]: $(U, (R, \varphi)) \to (U, (u, v))$

In [40]:
```
Pdisk_to_Pdisk_cart.display()
```

Out[40]:
$$\begin{cases} u = R\cos(\varphi) \\ v = R\sin(\varphi) \end{cases}$$

In [41]:
```
Pdisk_to_Pdisk_cart.set_inverse(sqrt(u^2+v^2), atan2(v, u))
Pdisk_to_Pdisk_cart.inverse().display()
```

```
Check of the inverse coordinate transformation:
  R == R  *passed*
  ph == arctan2(R*sin(ph), R*cos(ph))  **failed**
  u == u  *passed*
  v == v  *passed*
NB: a failed report can reflect a mere lack of simplification.
```

Out[41]:
$$\begin{cases} R = \sqrt{u^2+v^2} \\ \varphi = \arctan(v, u) \end{cases}$$

The embedding of $\mathbb{H}^2$ in $\mathbb{R}^3$ associated with the Poincaré disk model is naturally defined as

In [42]:
```
Phi2 = H2.diff_map(R3, {(X_Pdisk_cart, X3): [u, v, 0]},
                   name='Phi_2', latex_name=r'\Phi_2')
Phi2.display()
```

Out[42]:
$$\begin{array}{llcl} \Phi_2 : & \mathbb{H}^2 & \longrightarrow & \mathbb{R}^3 \\ & (u, v) & \longmapsto & (X, Y, Z) = (u, v, 0) \end{array}$$

Let us use it to draw the Poincaré disk in $\mathbb{R}^3$:

In [43]:
```
graph_disk_uv = X_Pdisk_cart.plot(X3, mapping=Phi2, number_values=15)
show(graph_disk_uv, figsize=7)
```



On U, the change of coordinates $(r, \varphi) \to (u, v)$ is obtained by combining the changes $(r, \varphi) \to (R, \varphi)$ and $(R, \varphi) \to (u, v)$:

In [44]:
```
pol_to_Pdisk_cart = Pdisk_to_Pdisk_cart * pol_to_Pdisk
pol_to_Pdisk_cart
```

Out[44]: $(U, (r, \varphi)) \to (U, (u, v))$

In [45]:
```
pol_to_Pdisk_cart.display()
```

Out[45]:
$$\left\{\begin{array}{lcl} u & = & \frac{r \cos\left(\varphi\right)}{\sqrt{r^{2} + 1} + 1} \\ v & = & \frac{r \sin\left(\varphi\right)}{\sqrt{r^{2} + 1} + 1} \end{array}\right.$$

Still on U, the change of coordinates $(X, Y) \to (u, v)$ is obtained by combining the changes $(X, Y) \to (r, \varphi)$ with $(r, \varphi) \to (u, v)$:

In [46]:
```
hyp_to_Pdisk_cart_U = pol_to_Pdisk_cart * pol_to_hyp.inverse()
hyp_to_Pdisk_cart_U
```

Out[46]: $(U, (X, Y)) \to (U, (u, v))$

In [47]:
```
hyp_to_Pdisk_cart_U.display()
```

Out[47]:
$$\left\{\begin{array}{lcl} u & = & \frac{X}{\sqrt{X^{2} + Y^{2} + 1} + 1} \\ v & = & \frac{Y}{\sqrt{X^{2} + Y^{2} + 1} + 1} \end{array}\right.$$

We use the above expression to extend the change of coordinates $(X, Y) \to (u, v)$ from U to the whole manifold $\mathbb{H}^2$:

In [48]:
```
hyp_to_Pdisk_cart = X_hyp.transition_map(X_Pdisk_cart, hyp_to_Pdisk_cart_U(X,Y))
hyp_to_Pdisk_cart
```

Out[48]: $\left(\mathbb{H}^2, (X, Y)\right) \to \left(\mathbb{H}^2, (u, v)\right)$

In [49]:
```
hyp_to_Pdisk_cart.display()
```

Out[49]:
$$\left\{\begin{array}{lcl} u & = & \frac{X}{\sqrt{X^{2} + Y^{2} + 1} + 1} \\ v & = & \frac{Y}{\sqrt{X^{2} + Y^{2} + 1} + 1} \end{array}\right.$$

In [50]:
```
hyp_to_Pdisk_cart.set_inverse(2*u/(1-u^2-v^2), 2*v/(1-u^2-v^2))
hyp_to_Pdisk_cart.inverse().display()
```

```
Check of the inverse coordinate transformation:
  X == X  *passed*
  Y == Y  *passed*
  u == -2*u*abs(u^2 + v^2 - 1)/(u^4 + 2*u^2*v^2 + v^4 + (u^2 + v^2 - 1)*abs(u^2 + v^2
- 1) - 1)  **failed**
  v == -2*v*abs(u^2 + v^2 - 1)/(u^4 + 2*u^2*v^2 + v^4 + (u^2 + v^2 - 1)*abs(u^2 + v^2
- 1) - 1)  **failed**
NB: a failed report can reflect a mere lack of simplification.
```

Out[50]:
$$\left\{\begin{array}{lcl} X & = & -\frac{2 \, u}{u^{2} + v^{2} - 1} \\ Y & = & -\frac{2 \, v}{u^{2} + v^{2} - 1} \end{array}\right.$$

In [51]:
```
graph_Pdisk = X_pol.plot(X3, mapping=Phi2.restrict(U), ranges={r: (0, 20)}, number_va
lues=15,
                          label_axes=False)
show(graph_hyp + graph_Pdisk, aspect_ratio=1, figsize=7)
```



In [52]:
```
X_pol.plot(X_Pdisk_cart, ranges={r: (0, 20)}, number_values=15)
```

Out[52]:



Metric tensor in Poincaré disk coordinates (u, v)

From now on, we are using the Poincaré disk chart $(\mathbb{H}^2, (u, v))$ as the default one on $\mathbb{H}^2$:

In [53]:
```
H2.set_default_chart(X_Pdisk_cart)
H2.set_default_frame(X_Pdisk_cart.frame())
```

In [54]:
```
g.display(X_hyp.frame())
```

Out[54]:

$$g = \left(\frac{u^4 + v^4 + 2\left(u^2+1\right)v^2 - 2u^2 + 1}{u^4 + v^4 + 2\left(u^2+1\right)v^2 + 2u^2 + 1}\right) \mathrm{d}X \otimes \mathrm{d}X + \left(-\frac{4uv}{u^4 + v^4 + 2\left(u^2+1\right)v^2 + 2u^2 + 1}\right) \mathrm{d}X \otimes$$
$$+ \left(-\frac{4uv}{u^4 + v^4 + 2\left(u^2+1\right)v^2 + 2u^2 + 1}\right) \mathrm{d}Y \otimes \mathrm{d}X + \left(\frac{u^4 + v^4 + 2\left(u^2-1\right)v^2 + 2u^2 + 1}{u^4 + v^4 + 2\left(u^2+1\right)v^2 + 2u^2 + 1}\right) \mathrm{d}Y \otimes \mathrm{d}$$

In [55]:
```
g.display()
```

Out[55]:

$$g = \left(\frac{4}{u^4 + v^4 + 2\left(u^2-1\right)v^2 - 2u^2 + 1}\right) \mathrm{d}u \otimes \mathrm{d}u + \left(\frac{4}{u^4 + v^4 + 2\left(u^2-1\right)v^2 - 2u^2 + 1}\right) \mathrm{d}v \otimes \mathrm{d}v$$

In [56]:
```
g[1,1].factor() ; g[2,2].factor()
g.display()
```

Out[56]:

$$g = \frac{4}{\left(u^2+v^2-1\right)^2} \mathrm{d}u \otimes \mathrm{d}u + \frac{4}{\left(u^2+v^2-1\right)^2} \mathrm{d}v \otimes \mathrm{d}v$$

Hemispherical model

The **hemispherical model of** $\mathbb{H}^2$ is obtained by the inverse stereographic projection from the point $S = (0, 0, -1)$ of the Poincaré disk to the unit sphere $X^2 + Y^2 + Z^2 = 1$. This induces a spherical coordinate chart on U:

In [57]:
```
X_spher.<th,ph> = U.chart(r'th:(0,pi/2):\theta ph:(0,2*pi):\varphi')
X_spher
```

Out[57]: $(U, (\theta, \varphi))$

From the stereographic projection from S, we obtain that

$$\sin\theta = \frac{2R}{1+R^2}$$

Hence the transition map:

In [58]:
```
Pdisk_to_spher = X_Pdisk.transition_map(X_spher, [arcsin(2*R/(1+R^2)), ph])
Pdisk_to_spher
```

Out[58]: $(U, (R, \varphi)) \rightarrow (U, (\theta, \varphi))$

In [59]:
```
Pdisk_to_spher.display()
```

Out[59]:

$$\left\{\begin{array}{lcl} \theta & = & \arcsin\left(\frac{2R}{R^2+1}\right) \\ \varphi & = & \varphi \end{array}\right.$$

In [60]:
```
Pdisk_to_spher.set_inverse(sin(th)/(1+cos(th)), ph)
Pdisk_to_spher.inverse().display()
```

Out[60]:

$$\left\{\begin{array}{lcl} R & = & \frac{\sin(\theta)}{\cos(\theta)+1} \\ \varphi & = & \varphi \end{array}\right.$$

In the spherical coordinates (θ, φ), the metric takes the following form:

In [61]:
```
g.display(X_spher.frame(), X_spher)
```

Out[61]:

$$g = \frac{1}{\cos(\theta)^2} \mathrm{d}\theta \otimes \mathrm{d}\theta + \frac{\sin(\theta)^2}{\cos(\theta)^2} \mathrm{d}\varphi \otimes \mathrm{d}\varphi$$

The embedding of $\mathbb{H}^2$ in $\mathbb{R}^3$ associated with the hemispherical model is naturally:

In [62]:
```
Phi3 = H2.diff_map(R3, {(X_spher, X3): [sin(th)*cos(ph), sin(th)*sin(ph), cos(th)]},
                   name='Phi_3', latex_name=r'\Phi_3')
Phi3.display()
```

Out[62]:
$$\begin{array}{llcl} \Phi_3: & \mathbb{H}^2 & \longrightarrow & \mathbb{R}^3 \\ \text{on } U: & (R,\varphi) & \longmapsto & (X,Y,Z) = \left(\frac{2\,R\cos(\varphi)}{R^2+1}, \frac{2\,R\sin(\varphi)}{R^2+1}, -\frac{R^2-1}{R^2+1}\right) \\ \text{on } U: & (\theta,\varphi) & \longmapsto & (X,Y,Z) = (\cos(\varphi)\sin(\theta), \sin(\varphi)\sin(\theta), \cos(\theta)) \end{array}$$

In [63]:
```
graph_spher = X_pol.plot(X3, mapping=Phi3, ranges={r: (0, 20)}, number_values=15,
                         color='orange', label_axes=False)
show(graph_hyp + graph_Pdisk + graph_spher, aspect_ratio=1,
     figsize=7)
```



Poincaré half-plane model

The **Poincaré half-plane model of** $\mathbb{H}^2$ is obtained by stereographic projection from the point $W = (-1, 0, 0)$ of the hemispherical model to the plane $X = 1$. This induces a new coordinate chart on $\mathbb{H}^2$ by setting $(x, y) = (Y, Z)$ in the plane $X = 1$:

In [64]:
```
X_hplane.<x,y> = H2.chart('x y:(0,+oo)')
X_hplane
```

Out[64]: $(\mathbb{H}^2, (x, y))$

The coordinate transformation $(\theta, \varphi) \to (x, y)$ is easily deduced from the stereographic projection from the point W:

In [65]:
```
spher_to_hplane = X_spher.transition_map(X_hplane, [2*sin(th)*sin(ph)/(1+sin(th)*cos
(ph)),
                                                    2*cos(th)/(1+sin(th)*cos(ph))])
spher_to_hplane
```

Out[65]: $(U, (\theta, \varphi)) \to (U, (x, y))$

In [66]:
```
spher_to_hplane.display()
```

Out[66]:
$$\left\{\begin{array}{lcl} x & = & \frac{2\,\sin(\varphi)\sin(\theta)}{\cos(\varphi)\sin(\theta)+1} \\ y & = & \frac{2\,\cos(\theta)}{\cos(\varphi)\sin(\theta)+1} \end{array}\right.$$

In [67]:
```
Pdisk_to_hplane = spher_to_hplane * Pdisk_to_spher
Pdisk_to_hplane
```

Out[67]: $(U,(R,\varphi)) \to (U,(x,y))$

In [68]:
```
Pdisk_to_hplane.display()
```

Out[68]:
$$\begin{cases} x & = & \frac{4\,R\,\sin(\varphi)}{R^2+2\,R\,\cos(\varphi)+1} \\ y & = & -\frac{2\,(R^2-1)}{R^2+2\,R\,\cos(\varphi)+1} \end{cases}$$

In [69]:
```
Pdisk_cart_to_hplane_U = Pdisk_to_hplane * Pdisk_to_Pdisk_cart.inverse()
Pdisk_cart_to_hplane_U
```

Out[69]: $(U,(u,v)) \to (U,(x,y))$

In [70]:
```
Pdisk_cart_to_hplane_U.display()
```

Out[70]:
$$\begin{cases} x & = & \frac{4\,v}{u^2+v^2+2\,u+1} \\ y & = & -\frac{2\,(u^2+v^2-1)}{u^2+v^2+2\,u+1} \end{cases}$$

Let us use the above formula to define the transition map $(u,v) \to (x,y)$ on the whole manifold $\mathbb{H}^2$ (and not only on U):

In [71]:
```
Pdisk_cart_to_hplane = X_Pdisk_cart.transition_map(X_hplane, Pdisk_cart_to_hplane_U
(u,v))
Pdisk_cart_to_hplane
```

Out[71]: $\left(\mathbb{H}^2,(u,v)\right) \to \left(\mathbb{H}^2,(x,y)\right)$

In [72]:
```
Pdisk_cart_to_hplane.display()
```

Out[72]:
$$\begin{cases} x & = & \frac{4\,v}{u^2+v^2+2\,u+1} \\ y & = & -\frac{2\,(u^2+v^2-1)}{u^2+v^2+2\,u+1} \end{cases}$$

In [73]:
```
Pdisk_cart_to_hplane.set_inverse((4-x^2-y^2)/(x^2+(2+y)^2), 4*x/(x^2+(2+y)^2))
Pdisk_cart_to_hplane.inverse().display()
```

Out[73]:
$$\begin{cases} u & = & -\frac{x^2+y^2-4}{x^2+(y+2)^2} \\ v & = & \frac{4\,x}{x^2+(y+2)^2} \end{cases}$$

Since the coordinates (x,y) correspond to (Y,Z) in the plane $X=1$, the embedding of $\mathbb{H}^2$ in $\mathbb{R}^3$ naturally associated with the Poincaré half-plane model is

In [74]:
```
Phi4 = H2.diff_map(R3, {(X_hplane, X3): [1, x, y]}, name='Phi_4', latex_name=r'\Phi_4
')
Phi4.display()
```

Out[74]:
$$\begin{array}{llcl} \Phi_4: & \mathbb{H}^2 & \longrightarrow & \mathbb{R}^3 \\ & (u,v) & \longmapsto & (X,Y,Z) = \left(1, \frac{4\,v}{u^2+v^2+2\,u+1}, -\frac{2\,(u^2+v^2-1)}{u^2+v^2+2\,u+1}\right) \\ & (x,y) & \longmapsto & (X,Y,Z) = (1,x,y) \end{array}$$

In [75]:
```
graph_hplane = X_pol.plot(X3, mapping=Phi4.restrict(U), ranges={r: (0, 1.5)},
                          number_values=15, color='brown', label_axes=False)
show(graph_hyp + graph_Pdisk + graph_spher + graph_hplane,
     aspect_ratio=1, figsize=8)
```



Let us draw the grid of the hyperboloidal coordinates (r, φ) in terms of the half-plane coordinates (x, y):

In [76]:
```
pol_to_hplane = Pdisk_to_hplane * pol_to_Pdisk
```

In [77]:
```
show(X_pol.plot(X_hplane, ranges={r: (0,24)}, style={r: '-', ph: '--'}, number_values
=15,
                plot_points=200, color='brown'), xmin=-5, xmax=5, ymin=0, ymax=5,
     aspect_ratio=1)
```



The solid curves are those along which r varies while φ is kept constant. Conversely, the dashed curves are those along which φ varies, while r is kept constant. We notice that the former curves are arcs of circles orthogonal to the half-plane boundary $y = 0$, hence they are geodesics of $(\mathbb{H}^2, g)$. This is not surprising since they correspond to the intersections of the hyperboloid with planes through the origin (namely the plane $\varphi = \mathrm{const}$). The point $(x, y) = (0, 2)$ corresponds to $r = 0$.

We may also depict the Poincaré disk coordinates (u, v) in terms of the half-plane coordinates (x, y):

In [78]:
```
X_Pdisk_cart.plot(ranges={u: (-1, 0), v: (-1, 0)},
                  style={u: '-', v: '--'}) + \
X_Pdisk_cart.plot(ranges={u: (-1, 0), v: (0., 1)},
                  style={u: '-', v: '--'}, color='orange') + \
X_Pdisk_cart.plot(ranges={u: (0, 1), v: (-1, 0)},
                    style={u: '-', v: '--'}, color='pink') + \
X_Pdisk_cart.plot(ranges={u: (0, 1), v: (0, 1)},
                  style={u: '-', v: '--'}, color='violet')
```

Out[78]:



In [79]:
```
show(X_Pdisk_cart.plot(X_hplane, ranges={u: (-1, 0), v: (-1, 0)},
                       style={u: '-', v: '--'}) + \
     X_Pdisk_cart.plot(X_hplane, ranges={u: (-1, 0), v: (0, 1)},
                       style={u: '-', v: '--'}, color='orange') + \
     X_Pdisk_cart.plot(X_hplane, ranges={u: (0, 1), v: (-1, 0)},
                       style={u: '-', v: '--'}, color='pink') + \
     X_Pdisk_cart.plot(X_hplane, ranges={u: (0, 1), v: (0, 1)},
                       style={u: '-', v: '--'}, color='violet'),
     xmin=-5, xmax=5, ymin=0, ymax=5, aspect_ratio=1)
```



The expression of the metric tensor in the half-plane coordinates (x, y) is

In [80]:
```
g.display(X_hplane.frame(), X_hplane)
```

Out[80]:

$$g = \frac{1}{y^2}\mathrm{d}x \otimes \mathrm{d}x + \frac{1}{y^2}\mathrm{d}y \otimes \mathrm{d}y$$

Summary

9 charts have been defined on $\mathbb{H}^2$:

In [81]:
```
H2.atlas()
```

Out[81]: $\left[\left(\mathbb{H}^2,(X,Y)\right),(U,(X,Y)),(U,(r,\varphi)),(U,(R,\varphi)),\left(\mathbb{H}^2,(u,v)\right),(U,(u,v)),(U,(\theta,\varphi)),\left(\mathbb{H}^2,(x,y)\right),(U,\right.$

There are actually 6 main charts, the other ones being subcharts:

In [82]:
```
H2.top_charts()
```

Out[82]: $\left[\left(\mathbb{H}^2,(X,Y)\right),(U,(r,\varphi)),(U,(R,\varphi)),\left(\mathbb{H}^2,(u,v)\right),(U,(\theta,\varphi)),\left(\mathbb{H}^2,(x,y)\right)\right]$

The expression of the metric tensor in each of these charts is

In [83]:
```
for chart in H2.top_charts():
    show(g.display(chart.frame(), chart))
```

$$g = \left(\frac{Y^2+1}{X^2+Y^2+1}\right)\mathrm{d}X\otimes\mathrm{d}X + \left(-\frac{XY}{X^2+Y^2+1}\right)\mathrm{d}X\otimes\mathrm{d}Y + \left(-\frac{XY}{X^2+Y^2+1}\right)\mathrm{d}Y\otimes\mathrm{d}X + \left(\frac{}{X}\right.$$

$$g = \left(\frac{1}{r^2+1}\right)\mathrm{d}r\otimes\mathrm{d}r + r^2\mathrm{d}\varphi\otimes\mathrm{d}\varphi$$

$$g = \frac{4}{(R+1)^2(R-1)^2}\mathrm{d}R\otimes\mathrm{d}R + \frac{4\,R^2}{(R+1)^2(R-1)^2}\mathrm{d}\varphi\otimes\mathrm{d}\varphi$$

$$g = \frac{4}{\left(u^2+v^2-1\right)^2}\mathrm{d}u\otimes\mathrm{d}u + \frac{4}{\left(u^2+v^2-1\right)^2}\mathrm{d}v\otimes\mathrm{d}v$$

$$g = \frac{1}{\cos(\theta)^2}\mathrm{d}\theta\otimes\mathrm{d}\theta + \frac{\sin(\theta)^2}{\cos(\theta)^2}\mathrm{d}\varphi\otimes\mathrm{d}\varphi$$

$$g = \frac{1}{y^2}\mathrm{d}x\otimes\mathrm{d}x + \frac{1}{y^2}\mathrm{d}y\otimes\mathrm{d}y$$

For each of these charts, the non-vanishing (and non-redundant w.r.t. the symmetry on the last 2 indices) **Christoffel symbols of** g are

In [84]:
```
for chart in H2.top_charts():
    show(chart)
    show(g.christoffel_symbols_display(chart=chart))
```

$\left(\mathbb{H}^2,(X, Y)\right)$

$$\begin{array}{lcl} \Gamma_{\,X\,X}^{\,X} & = & -\frac{XY^2+X}{X^2+Y^2+1} \\ \Gamma_{\,X\,Y}^{\,X} & = & \frac{X^2Y}{X^2+Y^2+1} \\ \Gamma_{\,Y\,Y}^{\,X} & = & -\frac{X^3+X}{X^2+Y^2+1} \\ \Gamma_{\,X\,X}^{\,Y} & = & -\frac{Y^3+Y}{X^2+Y^2+1} \\ \Gamma_{\,X\,Y}^{\,Y} & = & \frac{XY^2}{X^2+Y^2+1} \\ \Gamma_{\,Y\,Y}^{\,Y} & = & -\frac{(X^2+1)Y}{X^2+Y^2+1} \end{array}$$

$(U,(r, \varphi))$

$$\begin{array}{lcl} \Gamma_{\,r\,r}^{\,r} & = & -\frac{r}{r^2+1} \\ \Gamma_{\,\varphi\,\varphi}^{\,r} & = & -r^3 - r \\ \Gamma_{\,r\,\varphi}^{\,\varphi} & = & \frac{1}{r} \end{array}$$

$(U,(R, \varphi))$

$$\begin{array}{lcl} \Gamma_{\,R\,R}^{\,R} & = & -\frac{2\,R}{R^2-1} \\ \Gamma_{\,\varphi\,\varphi}^{\,R} & = & \frac{R^3+R}{R^2-1} \\ \Gamma_{\,R\,\varphi}^{\,\varphi} & = & -\frac{R^2+1}{R^3-R} \end{array}$$

$\left(\mathbb{H}^2,(u, v)\right)$

$$\begin{array}{lcl} \Gamma_{\,u\,u}^{\,u} & = & -\frac{2\,u}{u^2+v^2-1} \\ \Gamma_{\,u\,v}^{\,u} & = & -\frac{2\,v}{u^2+v^2-1} \\ \Gamma_{\,v\,v}^{\,u} & = & \frac{2\,u}{u^2+v^2-1} \\ \Gamma_{\,u\,u}^{\,v} & = & \frac{2\,v}{u^2+v^2-1} \\ \Gamma_{\,u\,v}^{\,v} & = & -\frac{2\,u}{u^2+v^2-1} \\ \Gamma_{\,v\,v}^{\,v} & = & -\frac{2\,v}{u^2+v^2-1} \end{array}$$

$(U,(\theta, \varphi))$

$$\begin{array}{lcl} \Gamma_{\,\theta\,\theta}^{\,\theta} & = & \frac{\sin(\theta)}{\cos(\theta)} \\ \Gamma_{\,\varphi\,\varphi}^{\,\theta} & = & -\frac{\sin(\theta)}{\cos(\theta)} \\ \Gamma_{\,\theta\,\varphi}^{\,\varphi} & = & \frac{1}{\cos(\theta)\sin(\theta)} \end{array}$$

$\left(\mathbb{H}^2,(x, y)\right)$

$$\begin{array}{lcl} \Gamma_{\,x\,y}^{\,x} & = & -\frac{1}{y} \\ \Gamma_{\,x\,x}^{\,y} & = & \frac{1}{y} \\ \Gamma_{\,y\,y}^{\,y} & = & -\frac{1}{y} \end{array}$$