

Sphere \mathbb{S}^2

This notebook demonstrates some differential geometry capabilities of SageMath on the example of the 2-dimensional sphere. The corresponding tools have been developed within the [SageManifolds](http://sagemanifolds.obspm.fr) (<http://sagemanifolds.obspm.fr>) project.

Click [here](https://raw.githubusercontent.com/sagemanifolds/SageManifolds/master/Notebooks/SM_sphere_S2.ipynb) (https://raw.githubusercontent.com/sagemanifolds/SageManifolds/master/Notebooks/SM_sphere_S2.ipynb) to download the notebook file (ipynb format). To run it, you must start SageMath with the Jupyter interface, via the command `sage -n jupyter`

NB: a version of SageMath at least equal to 7.5 is required to run this notebook:

Entrée [1]: `version()`

Out[1]: 'SageMath version 8.8, Release Date: 2019-06-26'

First we set up the notebook to display math formulas using LaTeX formatting:

Entrée [2]: `%display latex`

We also define a viewer for 3D plots (use `'threejs'` or `'jmol'` for interactive 3D graphics):

Entrée [3]: `viewer3D = 'threejs' # must be 'threejs', 'jmol', 'tachyon' or None (default)`

\mathbb{S}^2 as a 2-dimensional differentiable manifold

We start by declaring \mathbb{S}^2 as a differentiable manifold of dimension 2 over \mathbb{R} :

Entrée [4]: `S2 = Manifold(2, 'S^2', latex_name=r'\mathbb{S}^2', start_index=1)`

The first argument, 2, is the dimension of the manifold, while the second argument is the symbol used to label the manifold.

The argument `start_index` sets the index range to be used on the manifold for labelling components w.r.t. a basis or a frame: `start_index=1` corresponds to $\{1, 2\}$; the default value is `start_index=0` and yields to $\{0, 1\}$.

Entrée [5]: `print(S2)`

2-dimensional differentiable manifold S^2

Entrée [6]: `S2`

Out[6]: \mathbb{S}^2

The manifold is a `Parent` object:

Entrée [7]: `isinstance(S2, Parent)`

Out[7]: True

in the category of smooth manifolds over \mathbb{R} :

Entrée [8]: `S2.category()`

Out[8]: $\text{Smooth}_{\mathbb{R}}$

Coordinate charts on \mathbb{S}^2

The sphere cannot be covered by a single chart. At least two charts are necessary, for instance the charts associated with the stereographic projections from the North pole and the South pole respectively. Let us introduce the open subsets covered by these two charts:

$$U := \mathbb{S}^2 \setminus \{N\},$$

$$V := \mathbb{S}^2 \setminus \{S\},$$

where N is a point of \mathbb{S}^2 , which we shall call the *North pole*, and S is the point of U of stereographic coordinates $(0, 0)$, which we call the *South pole*:

Entrée [9]: `U = S2.open_subset('U') ; print(U)`

Open subset U of the 2-dimensional differentiable manifold S^2

Entrée [10]: `V = S2.open_subset('V') ; print(V)`

Open subset V of the 2-dimensional differentiable manifold S^2

We declare that $\mathbb{S}^2 = U \cup V$:

Entrée [11]: `S2.declare_union(U, V)`

Then we declare the stereographic chart on U , denoting by (x, y) the coordinates resulting from the stereographic projection from the North pole:

Entrée [12]: `stereoN.<x,y> = U.chart()`

The expression `.<x,y>` in the left-hand side means that the Python variables `x` and `y` are set to the two coordinates of the chart. This allows one to refer subsequently to the coordinates by their names. In the present case, the function `chart()` has no argument, which implies that the coordinate symbols will be `x` and `y` (i.e. exactly the characters appearing in the `<...>` operator) and that each coordinate range is $(-\infty, +\infty)$. As we will see below, for other cases, an argument must be passed to `chart()` to specify each coordinate symbol and range, as well as some specific LaTeX symbol.

Entrée [13]: `stereoN`

Out[13]: $(U, (x, y))$

The coordinates can be accessed individually, either by means of their indices in the chart (following the convention `start_index=1` set in the manifold's definition) or by their names as Python variables:

Entrée [14]: `stereoN[1]`

Out[14]: x

Entrée [15]: `y is stereoN[2]`

Out[15]: `True`

Similarly, we introduce on V the coordinates (x', y') corresponding to the stereographic projection from the South pole:

Entrée [16]: `stereoS.<xp,yp> = V.chart("xp:x' yp:y'")`

In this case, the string argument passed to `chart` stipulates that the text-only names of the coordinates are `xp` and `yp` (same as the Python variables names defined within the `<...>` operator in the left-hand side), while their LaTeX names are x' and y' .

Entrée [17]: `stereoS`

Out[17]: $(V, (x', y'))$

At this stage, the user's atlas on the manifold has two charts:

Entrée [18]: `S2.atlas()`

Out[18]: $[(U, (x, y)), (V, (x', y'))]$

We have to specify the **transition map** between the charts 'stereoN' = $(U, (x, y))$ and 'stereoS' = $(V, (x', y'))$; it is given by the standard inversion formulas:

Entrée [19]: `stereoN_to_S = stereoN.transition_map(stereoS,
 (x/(x^2+y^2), y/(x^2+y^2)),
 intersection_name='W',
 restrictions1= x^2+y^2!=0,
 restrictions2= xp^2+yp^2!=0)
 stereoN_to_S.display()`

Out[19]:
$$\begin{cases} x' &= \frac{x}{x^2+y^2} \\ y' &= \frac{y}{x^2+y^2} \end{cases}$$

In the above declaration, 'W' is the name given to the chart-overlap subset: $W := U \cap V$, the condition $x^2 + y^2 \neq 0$ defines W as a subset of U , and the condition $x'^2 + y'^2 \neq 0$ defines W as a subset of V .

The inverse coordinate transformation is computed by means of the method `inverse()` :

Entrée [20]: `stereoS_to_N = stereoN_to_S.inverse()
 stereoS_to_N.display()`

Out[20]:
$$\begin{cases} x &= \frac{x'}{x'^2+y'^2} \\ y &= \frac{y'}{x'^2+y'^2} \end{cases}$$

In the present case, the situation is of course perfectly symmetric regarding the coordinates (x, y) and (x', y') .

At this stage, the user's atlas has four charts:

Entrée [21]: `S2.atlas()`

Out[21]: $[(U, (x, y)), (V, (x', y')), (W, (x, y)), (W, (x', y'))]$

Let us store $W = U \cap V$ into a Python variable for future use:

Entrée [22]: `W = U.intersection(V)`

Similarly we store the charts $(W, (x, y))$ (the restriction of $(U, (x, y))$ to W) and $(W, (x', y'))$ (the restriction of $(V, (x', y'))$ to W) into Python variables:

Entrée [23]: `stereoN_W = stereoN.restrict(W)
 stereoN_W`

Out[23]: $(W, (x, y))$


Entrée [24]: `stereoS_W = stereoS.restrict(W)
 stereoS_W`

Out[24]: $(W, (x', y'))$

We may plot the chart $(W, (x', y'))$ in terms of itself, as a grid:

Entrée [25]: `stereoS_W.plot()`


Out[25]:



More interestingly, let us plot the stereographic chart (x', y') in terms of the stereographic chart (x, y) on the domain W where both systems overlap (we split the plot in four parts to avoid the singularity at $(x', y') = (0, 0)$):

Entrée [26]:

```
graphSN1 = stereoS_W.plot(stereoN, ranges={xp: [-6, -0.02], yp: [-6, -0.02]})
graphSN2 = stereoS_W.plot(stereoN, ranges={xp: [-6, -0.02], yp: [0.02, 6]})
graphSN3 = stereoS_W.plot(stereoN, ranges={xp: [0.02, 6], yp: [-6, -0.02]})
graphSN4 = stereoS_W.plot(stereoN, ranges={xp: [0.02, 6], yp: [0.02, 6]})
show(graphSN1+graphSN2+graphSN3+graphSN4,
      xmin=-1.5, xmax=1.5, ymin=-1.5, ymax=1.5)
```



Spherical coordinates

The standard spherical (or polar) coordinates (θ, ϕ) are defined on the open domain $A \subset W \subset \mathbb{S}^2$ that is the complement of the "origin meridian"; since the latter is the half-circle defined by $y = 0$ and $x \geq 0$, we declare:

```
Entrée [27]: A = W.open_subset('A', coord_def={stereoN_W: (y!=0, x<0),
                                                stereoS_W: (yp!=0, xp<0)})
print(A)
```

Open subset A of the 2-dimensional differentiable manifold S^2

The restriction of the stereographic chart from the North pole to A is

```
Entrée [28]: stereoN_A = stereoN_W.restrict(A)
stereoN_A
```

```
Out[28]: (A, (x, y))
```

We then declare the chart $(A, (\theta, \phi))$ by specifying the intervals $(0, \pi)$ and $(0, 2\pi)$ spanned by respectively θ and ϕ :

```
Entrée [29]: sphr.<th,ph> = A.chart(r'th:(0,pi):\theta ph:(0,2*pi):\phi') ; sphr
```

```
Out[29]: (A, (\theta, \phi))
```

The specification of the spherical coordinates is completed by providing the transition map with the stereographic chart $(A, (x, y))$:

```
Entrée [30]: sphr_to_stereoN = sphr.transition_map(stereoN_A,
                                                    (sin(th)*cos(ph)/(1-cos(th)),
                                                     sin(th)*sin(ph)/(1-cos(th))))
sphr_to_stereoN.display()
```

```
Out[30]: { x = -\frac{\cos(\phi) \sin(\theta)}{\cos(\theta)-1}
          y = -\frac{\sin(\phi) \sin(\theta)}{\cos(\theta)-1}
```

We also provide the inverse transition map, asking to check that the provided formulas are indeed correct (argument verbose=True):

```
Entrée [31]: sphr_to_stereoN.set_inverse(2*atan(1/sqrt(x^2+y^2)), atan2(-y,-x)+pi,
                                         verbose=True)
```

```
Check of the inverse coordinate transformation:
th == 2*arctan(sqrt(-cos(th) + 1)/sqrt(cos(th) + 1))
ph == pi + arctan2(sin(ph)*sin(th)/(cos(th) - 1), cos(ph)*sin(th)/(cos(th) - 1))
x == x
y == y
```

The check is passed, modulo some lack of trigonometric simplifications in the first two lines.

```
Entrée [32]: sphr_to_stereoN.inverse().display()
```

```
Out[32]: { \theta = 2 \arctan\left(\frac{1}{\sqrt{x^2+y^2}}\right)
          \phi = \pi + \arctan(-y, -x)
```

The transition map $(A, (\theta, \phi)) \rightarrow (A, (x', y'))$ is obtained by combining the transition maps $(A, (\theta, \phi)) \rightarrow (A, (x, y))$ and $(A, (x, y)) \rightarrow (A, (x', y'))$:

```
Entrée [33]: stereoN_to_S_A = stereoN_to_S.restrict(A)
sphr_to_stereoS = stereoN_to_S_A * sphr_to_stereoN
sphr_to_stereoS.display()
```

```
Out[33]: { x' = -\frac{\cos(\phi) \cos(\theta)-\cos(\phi)}{\sin(\theta)}
          y' = -\frac{\cos(\theta) \sin(\phi)-\sin(\phi)}{\sin(\theta)}
```

Similarly, the transition map $(A, (x', y')) \rightarrow (A, (\theta, \phi))$ is obtained by combining the transition maps $(A, (x', y')) \rightarrow (A, (x, y))$ and $(A, (x, y)) \rightarrow (A, (\theta, \phi))$:

```
Entrée [34]: stereoS_to_N_A = stereoN_to_S.inverse().restrict(A)
stereoS_to_spher = spher_to_stereoN.inverse() * stereoS_to_N_A
stereoS_to_spher.display()
```

Out[34]:
$$\begin{cases} \theta &= 2 \arctan\left(\sqrt{x'^2 + y'^2}\right) \\ \phi &= \pi - \arctan\left(\frac{y'}{x'^2 + y'^2}, -\frac{x'}{x'^2 + y'^2}\right) \end{cases}$$

The user atlas of \mathbb{S}^2 is now


```
Entrée [35]: S2.atlas()
```

Out[35]: $[(U, (x, y)), (V, (x', y')), (W, (x, y)), (W, (x', y')), (A, (x, y)), (A, (x', y')), (A, (\theta, \phi))]$

Let us draw the grid of spherical coordinates (θ, ϕ) in terms of stereographic coordinates from the North pole (x, y) :

```
Entrée [36]: spher.plot(stereoN, number_values=15, ranges={th: (pi/8, pi)})
```


Out[36]:



Conversely, we may represent the grid of the stereographic coordinates (x, y) restricted to A in terms of the spherical coordinates (θ, ϕ) . We limit ourselves to one quarter (cf. the argument `ranges`):

```
Entrée [37]: stereoN_A.plot(spher, ranges={x: (0.01, 8), y: (0.01, 8)}, number_values=20, plot_points=200)
```

Out[37]:



Points on \mathbb{S}^2

We declare the **North pole** (resp. the **South pole**) as the point of coordinates $(0, 0)$ in the chart $(V, (x', y'))$ (resp. in the chart $(U, (x, y))$):

```
Entrée [38]: N = V.point((0,0), chart=stereoS, name='N') ; print(N)
             S = U.point((0,0), chart=stereoN, name='S') ; print(S)
```

Point N on the 2-dimensional differentiable manifold S^2
 Point S on the 2-dimensional differentiable manifold S^2

Since points are Sage `Element`'s, the corresponding `Parent` being the manifold subsets, an equivalent writing of the above declarations is

```
Entrée [39]: N = V((0,0), chart=stereoS, name='N') ; print(N)
             S = U((0,0), chart=stereoN, name='S') ; print(S)
```

Point N on the 2-dimensional differentiable manifold S^2
 Point S on the 2-dimensional differentiable manifold S^2

Moreover, since `stereoS` is the default chart on V and `stereoN` is the default one on U , their mentions can be omitted, so that the above can be shortened to

```
Entrée [40]: N = V((0,0), name='N') ; print(N)
             S = U((0,0), name='S') ; print(S)
```

Point N on the 2-dimensional differentiable manifold S^2
 Point S on the 2-dimensional differentiable manifold S^2

```
Entrée [41]: N.parent()
```

```
Out[41]: V
```

```
Entrée [42]: S.parent()
```

```
Out[42]: U
```

We have of course

```
Entrée [43]: N in V
```

```
Out[43]: True
```

```
Entrée [44]: N in S2
```

```
Out[44]: True
```

```
Entrée [45]: N in U
```

```
Out[45]: False
```

```
Entrée [46]: N in A
```

```
Out[46]: False
```

Let us introduce some point at the equator:

```
Entrée [47]: E = S2((0,1), chart=stereoN, name='E')
```

The point E is in the open subset A :

```
Entrée [48]: E in A
```

```
Out[48]: True
```

We may then ask for its spherical coordinates (θ, ϕ) :

Entrée [49]: `E.coord(spher)`

Out[49]: $\left(\frac{1}{2}\pi, \frac{1}{2}\pi\right)$

which is not possible for the point N :

Entrée [50]:

```
try:
    N.coord(spher)
except ValueError as exc:
    print('Error: ' + str(exc))
```

Error: the point does not belong to the domain of Chart (A, (th, ph))

Maps between manifolds: the embedding of \mathbb{S}^2 into \mathbb{R}^3

Let us first declare \mathbb{R}^3 as a 3-dimensional manifold covered by a single chart (the so-called **Cartesian coordinates**):

Entrée [51]:

```
R3 = Manifold(3, 'R^3', r'\mathbb{R}^3', start_index=1)
cart.<X,Y,Z> = R3.chart() ; cart
```

Out[51]: $(\mathbb{R}^3, (X, Y, Z))$

The embedding of the sphere is defined as a differential map $\Phi : \mathbb{S}^2 \rightarrow \mathbb{R}^3$:

Entrée [52]:

```
Phi = S2.diff_map(R3, {(stereoN, cart):
    [2*x/(1+x^2+y^2), 2*y/(1+x^2+y^2),
    (x^2+y^2-1)/(1+x^2+y^2)],
    (stereoS, cart):
    [2*xp/(1+xp^2+yp^2), 2*yp/(1+xp^2+yp^2),
    (1-xp^2-yp^2)/(1+xp^2+yp^2)]},
    name='Phi', latex_name=r'\Phi')
```

Entrée [53]: `Phi.display()`

Out[53]:
$$\begin{aligned} \Phi : \quad \mathbb{S}^2 &\longrightarrow \mathbb{R}^3 \\ \text{on } U : \quad (x, y) &\longmapsto (X, Y, Z) = \left(\frac{2x}{x^2+y^2+1}, \frac{2y}{x^2+y^2+1}, \frac{x^2+y^2-1}{x^2+y^2+1} \right) \\ \text{on } V : \quad (x', y') &\longmapsto (X, Y, Z) = \left(\frac{2x'}{x'^2+y'^2+1}, \frac{2y'}{x'^2+y'^2+1}, -\frac{x'^2+y'^2-1}{x'^2+y'^2+1} \right) \end{aligned}$$

Entrée [54]: `Phi.parent()`

Out[54]: $\text{Hom}(\mathbb{S}^2, \mathbb{R}^3)$

Entrée [55]: `print(Phi.parent())`

Set of Morphisms from 2-dimensional differentiable manifold S^2 to 3-dimensional differentiable manifold R^3 in Category of smooth manifolds over Real Field with 53 bits of precision

Entrée [56]: `Phi.parent() is Hom(S2, R3)`

Out[56]: True

Φ maps points of \mathbb{S}^2 to points of \mathbb{R}^3 :

Entrée [57]: `N1 = Phi(N) ; print(N1) ; N1 ; N1.coord()`

Point Phi(N) on the 3-dimensional differentiable manifold R^3

Out[57]: $(0, 0, 1)$

Entrée [58]: `S1 = Phi(S) ; print(S1) ; S1 ; S1.coord()`

Point Phi(S) on the 3-dimensional differentiable manifold \mathbb{R}^3

Out[58]: $(0, 0, -1)$

Entrée [59]: `E1 = Phi(E) ; print(E1) ; E1 ; E1.coord()`

Point Phi(E) on the 3-dimensional differentiable manifold \mathbb{R}^3

Out[59]: $(0, 1, 0)$

Φ has been defined in terms of the stereographic charts $(U, (x, y))$ and $(V, (x', y'))$, but we may ask its expression in terms of spherical coordinates. The latter is then computed by means of the transition map $(A, (x, y)) \rightarrow (A, (\theta, \phi))$:

Entrée [60]: `Phi.expr(stereoN_A, cart)`

Out[60]: $\left(\frac{2x}{x^2 + y^2 + 1}, \frac{2y}{x^2 + y^2 + 1}, \frac{x^2 + y^2 - 1}{x^2 + y^2 + 1} \right)$

Entrée [61]: `Phi.expr(spher, cart)`


Out[61]: $(\cos(\phi) \sin(\theta), \sin(\phi) \sin(\theta), \cos(\theta))$

Entrée [62]: `Phi.display(spher, cart)`

Out[62]: $\Phi : \mathbb{S}^2 \longrightarrow \mathbb{R}^3$
on $A : (\theta, \phi) \longmapsto (X, Y, Z) = (\cos(\phi) \sin(\theta), \sin(\phi) \sin(\theta), \cos(\theta))$


Let us use Φ to draw the grid of spherical coordinates (θ, ϕ) in terms of the Cartesian coordinates (X, Y, Z) of \mathbb{R}^3 :

Entrée [63]: `graph_spher = spher.plot(chart=cart, mapping=Phi, number_values=11,
color='blue', label_axes=False)
show(graph_spher, viewer=viewer3D)`




We may also use the embedding Φ to display the stereographic coordinate grid in terms of the Cartesian coordinates in \mathbb{R}^3 . First for the stereographic coordinates from the North pole:

```
Entrée [64]: graph_stereoN = stereoN.plot(chart=cart, mapping=Phi, number_values=25,
                                           label_axes=False)
show(graph_stereoN, viewer=viewer3D)
```




and then have a view with the stereographic coordinates from the South pole superposed (in green):

```
Entrée [65]: graph_stereoS = stereoS.plot(chart=cart, mapping=Phi, number_values=25,
                                           color='green', label_axes=False)
show(graph_stereoN + graph_stereoS, viewer=viewer3D)
```



We may also add the two poles to the graphic:

```
Entrée [66]: pointN = N.plot(chart=cart, mapping=Phi, color='red',
                        label_offset=0.05)
pointS = S.plot(chart=cart, mapping=Phi, color='green',
                label_offset=0.05)
show(graph_stereoN + graph_stereoS + pointN + pointS, viewer=viewer3D)
```



Tangent spaces

The tangent space to the manifold \mathbb{S}^2 at the point N is

```
Entrée [67]: T_N = S2.tangent_space(N)
print(T_N) ; T_N
```

Tangent space at Point N on the 2-dimensional differentiable manifold S^2

Out[67]: $T_N \mathbb{S}^2$

$T_N \mathbb{S}^2$ is a vector space over \mathbb{R} (represented here by Sage's symbolic ring SR):

```
Entrée [68]: print(T_N.category())
```

Category of finite dimensional vector spaces over Symbolic Ring

Its dimension equals the manifold's dimension:

```
Entrée [69]: dim(T_N)
```

Out[69]: 2

```
Entrée [70]: dim(T_N) == dim(S2)
```

Out[70]: True

$T_N \mathbb{S}^2$ is endowed with a basis inherited from the coordinate frame defined around N , namely the frame associated with the chart $(V, (x', y'))$:

```
Entrée [71]: T_N.bases()
```

Out[71]: $\left[\left(\frac{\partial}{\partial x'}, \frac{\partial}{\partial y'} \right) \right]$

$(V, (x', y'))$ is the only chart defined so far around the point N . If various charts have been defined around a point, then the tangent space at this point is automatically endowed with the bases inherited from the coordinate frames associated to all these charts. For instance, for the equator point E :

Entrée [72]: `T_E = S2.tangent_space(E)`
`print(T_E) ; T_E`

Tangent space at Point E on the 2-dimensional differentiable manifold S^2

Out[72]: $T_E \mathbb{S}^2$

Entrée [73]: `T_E.bases()`

Out[73]: $\left[\left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y} \right), \left(\frac{\partial}{\partial x'}, \frac{\partial}{\partial y'} \right), \left(\frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right]$

Entrée [74]: `T_E.default_basis()`

Out[74]: $\left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y} \right)$

An element of $T_E \mathbb{S}^2$:

Entrée [75]: `v = T_E((-3, 2), name='v')`
`print(v)`

Tangent vector v at Point E on the 2-dimensional differentiable manifold S^2

Entrée [76]: `v in T_E`

Out[76]: True

Entrée [77]: `v.parent()`

Out[77]: $T_E \mathbb{S}^2$

Entrée [78]: `v.display()`

Out[78]: $v = -3 \frac{\partial}{\partial x} + 2 \frac{\partial}{\partial y}$

Entrée [79]: `v.display(T_E.bases()[1])`

Out[79]: $v = -3 \frac{\partial}{\partial x'} - 2 \frac{\partial}{\partial y'}$

Entrée [80]: `v.display(T_E.bases()[2])`

Out[80]: $v = -2 \frac{\partial}{\partial \theta} + 3 \frac{\partial}{\partial \phi}$

Differential of a smooth mapping

The differential of the mapping Φ at the point E is

Entrée [81]: `dPhi_E = Phi.differential(E)`
`print(dPhi_E) ; dPhi_E`

Generic morphism:

From: Tangent space at Point E on the 2-dimensional differentiable manifold S^2

To: Tangent space at Point Phi(E) on the 3-dimensional differentiable manifold \mathbb{R}^3

Out[81]: $d\Phi_E$

Entrée [82]: `dPhi_E.domain()`

Out[82]: $T_E \mathbb{S}^2$

Entrée [83]: `dPhi_E.codomain()`

Out[83]: $T_{\Phi(E)} \mathbb{R}^3$

Entrée [84]: `dPhi_E.parent()`

Out[84]: $\text{Hom}(T_E \mathbb{S}^2, T_{\Phi(E)} \mathbb{R}^3)$

The image by $d\Phi_E$ of the vector $v \in T_E \mathbb{S}^2$ introduced above is

Entrée [85]: `dPhi_E(v)`

Out[85]: $d\Phi_E(v)$

Entrée [86]: `print(dPhi_E(v))`

Tangent vector $d\Phi_E(v)$ at Point $\Phi(E)$ on the 3-dimensional differentiable manifold \mathbb{R}^3

Entrée [87]: `dPhi_E(v) in R3.tangent_space(Phi(E))`

Out[87]: True

Entrée [88]: `dPhi_E(v).display()`

Out[88]: $d\Phi_E(v) = 3\frac{\partial}{\partial X} - 2\frac{\partial}{\partial Z}$

Algebra of scalar fields

The set $C^\infty(\mathbb{S}^2)$ of all smooth functions $\mathbb{S}^2 \rightarrow \mathbb{R}$ has naturally the structure of a commutative algebra over \mathbb{R} . $C^\infty(\mathbb{S}^2)$ is obtained via the method `scalar_field_algebra()` applied to the manifold \mathbb{S}^2 :

Entrée [89]: `CS = S2.scalar_field_algebra() ; CS`

Out[89]: $C^\infty(\mathbb{S}^2)$

Since the algebra internal product is the pointwise multiplication, it is clearly commutative, so that $C^\infty(\mathbb{S}^2)$ belongs to Sage's category of commutative algebras:

Entrée [90]: `CS.category()`

Out[90]: **CommutativeAlgebras_{SR}**

The base ring of the algebra $C^\infty(\mathbb{S}^2)$ is the field \mathbb{R} , which is represented here by Sage's Symbolic Ring (SR):

Entrée [91]: `CS.base_ring()`

Out[91]: SR

Elements of $C^\infty(\mathbb{S}^2)$ are of course (smooth) scalar fields:

Entrée [92]: `print(CS.an_element())`

Scalar field on the 2-dimensional differentiable manifold \mathbb{S}^2

This example element is the constant scalar field that takes the value 2:

Entrée [93]: `CS.an_element().display()`

Out[93]:
$$\begin{array}{lll} \mathbb{S}^2 & \longrightarrow & \mathbb{R} \\ \text{on } U : & (x, y) & \longmapsto 2 \\ \text{on } V : & (x', y') & \longmapsto 2 \\ \text{on } A : & (\theta, \phi) & \longmapsto 2 \end{array}$$

A specific element is the zero one:

```
Entrée [94]: f = CS.zero()
             print(f)
```

Scalar field zero on the 2-dimensional differentiable manifold S^2

Scalar fields map points of \mathbb{S}^2 to real numbers:

```
Entrée [95]: f(N), f(E), f(S)
```

```
Out[95]: (0,0,0)
```

```
Entrée [96]: f.display()
```

```
Out[96]: 0 :       $\mathbb{S}^2$        $\longrightarrow$   $\mathbb{R}$ 
          on  $U$  :  $(x,y)$        $\longmapsto$  0
          on  $V$  :  $(x',y')$     $\longmapsto$  0
          on  $A$  :  $(\theta,\phi)$     $\longmapsto$  0
```

Another specific element is the algebra unit element, i.e. the constant scalar field 1:

```
Entrée [97]: f = CS.one()
             print(f)
```

Scalar field 1 on the 2-dimensional differentiable manifold S^2

```
Entrée [98]: f(N), f(E), f(S)
```

```
Out[98]: (1,1,1)
```

```
Entrée [99]: f.display()
```

```
Out[99]: 1 :       $\mathbb{S}^2$        $\longrightarrow$   $\mathbb{R}$ 
          on  $U$  :  $(x,y)$        $\longmapsto$  1
          on  $V$  :  $(x',y')$     $\longmapsto$  1
          on  $A$  :  $(\theta,\phi)$     $\longmapsto$  1
```

Let us define a scalar field by its coordinate expression in the two stereographic charts:

```
Entrée [100]: f = CS({stereoN: pi - 2*atan(x^2+y^2), stereoS: 2*atan(xp^2+yp^2)})
              f.display()
```

```
Out[100]:       $\mathbb{S}^2$        $\longrightarrow$   $\mathbb{R}$ 
          on  $U$  :  $(x,y)$        $\longmapsto$   $\pi - 2 \arctan(x^2 + y^2)$ 
          on  $V$  :  $(x',y')$     $\longmapsto$   $2 \arctan(x'^2 + y'^2)$ 
          on  $A$  :  $(\theta,\phi)$     $\longmapsto$   $\pi + 2 \arctan\left(\frac{\cos(\theta)+1}{\cos(\theta)-1}\right)$ 
```

Instead of using `CS()` (i.e. the `Parent__call__` function), we may invoke the `scalar_field` method on the manifold to create f ; this allows to pass the name of the scalar field:

```
Entrée [101]: f = S2.scalar_field({stereoN: pi - 2*atan(x^2+y^2), stereoS: 2*atan(xp^2+yp^2)}, name='f')
              f.display()
```

```
Out[101]: f :       $\mathbb{S}^2$        $\longrightarrow$   $\mathbb{R}$ 
          on  $U$  :  $(x,y)$        $\longmapsto$   $\pi - 2 \arctan(x^2 + y^2)$ 
          on  $V$  :  $(x',y')$     $\longmapsto$   $2 \arctan(x'^2 + y'^2)$ 
          on  $A$  :  $(\theta,\phi)$     $\longmapsto$   $\pi + 2 \arctan\left(\frac{\cos(\theta)+1}{\cos(\theta)-1}\right)$ 
```

Entrée [102]: `f.parent()`

Out[102]: $C^\infty(\mathbb{S}^2)$

Internally, the various coordinate expressions of the scalar field are stored in the dictionary `_express`, whose keys are the charts:

Entrée [103]: `f._express`

Out[103]: $\left\{ (U, (x, y)) : \pi - 2 \arctan(x^2 + y^2), (A, (\theta, \phi)) : \pi + 2 \arctan\left(\frac{\cos(\theta) + 1}{\cos(\theta) - 1}\right), (A, (x, y)) : \pi - 2 \arctan\left(\frac{\cos(\theta) + 1}{\cos(\theta) - 1}\right), (A, (x, y)) : \pi - 2 \arctan\left(\frac{\cos(\theta) + 1}{\cos(\theta) - 1}\right) \right\}$

The expression in a specific chart is recovered by passing the chart as the argument of the method `display()`:

Entrée [104]: `f.display(stereoS)`

Out[104]:
$$\begin{aligned} f : \quad \mathbb{S}^2 &\longrightarrow \mathbb{R} \\ \text{on } V : \quad (x', y') &\longmapsto 2 \arctan(x'^2 + y'^2) \end{aligned}$$

Scalar fields map the manifold's points to real numbers:

Entrée [105]: `f(N)`

Out[105]: 0

Entrée [106]: `f(E)`

Out[106]: $\frac{1}{2} \pi$

Entrée [107]: `f(S)`

Out[107]: π

We may define the restrictions of f to the open subsets U and V :

Entrée [108]: `fU = f.restrict(U)`
`fU.display()`

Out[108]:
$$\begin{aligned} f : \quad U &\longrightarrow \mathbb{R} \\ (x, y) &\longmapsto \pi - 2 \arctan(x^2 + y^2) \\ \text{on } W : \quad (x', y') &\longmapsto 2 \arctan(x'^2 + y'^2) \\ \text{on } A : \quad (\theta, \phi) &\longmapsto \pi + 2 \arctan\left(\frac{\cos(\theta) + 1}{\cos(\theta) - 1}\right) \end{aligned}$$

Entrée [109]: `fV = f.restrict(V)`
`fV.display()`

Out[109]:
$$\begin{aligned} f : \quad V &\longrightarrow \mathbb{R} \\ (x', y') &\longmapsto 2 \arctan(x'^2 + y'^2) \\ \text{on } W : \quad (x, y) &\longmapsto \pi - 2 \arctan(x^2 + y^2) \\ \text{on } A : \quad (\theta, \phi) &\longmapsto \pi + 2 \arctan\left(\frac{\cos(\theta) + 1}{\cos(\theta) - 1}\right) \end{aligned}$$

Entrée [110]: `fU(E), fV(S)`

Out[110]: $\left(\frac{1}{2} \pi, \pi\right)$

Entrée [111]: `fU.parent()`

Out[111]: $C^\infty(U)$

Entrée [112]: `fV.parent()`

Out[112]: $C^\infty(V)$

Entrée [113]: `CU = U.scalar_field_algebra()
fU.parent() is CU`

Out[113]: True

A scalar field on \mathbb{S}^2 can be coerced to a scalar field on U , the coercion being simply the restriction:

Entrée [114]: `CU.has_coerce_map_from(CS)`

Out[114]: True

Entrée [115]: `fU == CU(f)`

Out[115]: True

The arithmetic of scalar fields:

Entrée [116]: `g = f*f - 2*f
g.display()`

Out[116]:
$$\begin{aligned} & \mathbb{S}^2 \longrightarrow \mathbb{R} \\ \text{on } U : (x, y) & \longmapsto -2\pi + \pi^2 - 4(\pi - 1) \arctan(x^2 + y^2) + 4 \arctan(x^2 + y^2)^2 \\ \text{on } V : (x', y') & \longmapsto 4 \arctan(x'^2 + y'^2)^2 - 4 \arctan(x'^2 + y'^2) \\ \text{on } A : (\theta, \phi) & \longmapsto -2\pi + \pi^2 + 4(\pi - 1) \arctan\left(\frac{\cos(\theta)+1}{\cos(\theta)-1}\right) + 4 \arctan\left(\frac{\cos(\theta)+1}{\cos(\theta)-1}\right)^2 \end{aligned}$$

Module of vector fields

The set $\mathfrak{X}(\mathbb{S}^2)$ of all smooth vector fields on \mathbb{S}^2 is a module over the algebra (ring) $C^\infty(\mathbb{S}^2)$. It is obtained by the method `vector_field_module()`:

Entrée [117]: `XS = S2.vector_field_module()
XS`

Out[117]: $\mathfrak{X}(\mathbb{S}^2)$

Entrée [118]: `print(XS)`

Module $X(S^2)$ of vector fields on the 2-dimensional differentiable manifold S^2

Entrée [119]: `XS.base_ring()`

Out[119]: $C^\infty(\mathbb{S}^2)$

Entrée [120]: `XS.category()`

Out[120]: **Modules** $C^\infty(\mathbb{S}^2)$

$\mathfrak{X}(\mathbb{S}^2)$ is not a free module:

Entrée [121]: `isinstance(XS, FiniteRankFreeModule)`

Out[121]: False

because \mathbb{S}^2 is not a parallelizable manifold:

Entrée [122]: `S2.is_manifestly_parallelizable()`

Out[122]: False

On the contrary, the set $\mathfrak{X}(U)$ of smooth vector fields on U is a free module:

Entrée [123]: `XU = U.vector_field_module()
isinstance(XU, FiniteRankFreeModule)`

Out[123]: True

because U is parallelizable:

Entrée [124]: `U.is_manifestly_parallelizable()`

Out[124]: True

Due to the introduction of the stereographic coordinates (x, y) on U , a basis has already been defined on the free module $\mathfrak{X}(U)$, namely the coordinate basis $(\partial/\partial x, \partial/\partial y)$:

Entrée [125]: `XU.print_bases()`

Bases defined on the Free module $X(U)$ of vector fields on the Open subset U of the 2-dimensional differentiable manifold S^2 :
- $(U, (\partial/\partial x, \partial/\partial y))$ (default basis)

Entrée [126]: `XU.default_basis()`

Out[126]: $\left(U, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y} \right) \right)$

Similarly

Entrée [127]: `XV = V.vector_field_module()
XV.default_basis()`

Out[127]: $\left(V, \left(\frac{\partial}{\partial x'}, \frac{\partial}{\partial y'} \right) \right)$

Entrée [128]: `eU = XU.default_basis()
eV = XV.default_basis()`

From the point of view of the open set U , eU is also the default vector frame:

Entrée [129]: `eU is U.default_frame()`

Out[129]: True

It is also the default vector frame on S^2 (although not defined on the whole S^2), for it is the first vector frame defined on an open subset of S^2 :

Entrée [130]: `eU is S2.default_frame()`

Out[130]: True

Entrée [131]: `eV is V.default_frame()`

Out[131]: True

Let us introduce a vector field on S^2 :

```
Entrée [132]: v = S2.vector_field(name='v')
              v[eU,:] = [1, -2]
              v.display(eU)
```

Out[132]:
$$v = \frac{\partial}{\partial x} - 2 \frac{\partial}{\partial y}$$

```
Entrée [133]: v.parent()
```

Out[133]: $\mathfrak{X}(\mathbb{S}^2)$

```
Entrée [134]: stereoSW = stereoS.restrict(W)
              eSW = stereoSW.frame()
              eSW
```

Out[134]:
$$\left(W, \left(\frac{\partial}{\partial x'}, \frac{\partial}{\partial y'} \right) \right)$$

```
Entrée [135]: vW = v.restrict(W)
              vW.display()
```

Out[135]:
$$v = \frac{\partial}{\partial x} - 2 \frac{\partial}{\partial y}$$

```
Entrée [136]: vW.parent()
```

Out[136]: $\mathfrak{X}(W)$

```
Entrée [137]: print(vW.parent())
```

Free module $X(W)$ of vector fields on the Open subset W of the 2-dimensional differentiable manifold S^2

```
Entrée [138]: vW.display(eSW)
```

Out[138]:
$$v = \left(-\frac{x^2 - 4xy - y^2}{x^4 + 2x^2y^2 + y^4} \right) \frac{\partial}{\partial x'} + \left(-\frac{2(x^2 + xy - y^2)}{x^4 + 2x^2y^2 + y^4} \right) \frac{\partial}{\partial y'}$$

```
Entrée [139]: vW.display(eSW, stereoSW)
```

Out[139]:
$$v = \left(-x'^2 + 4x'y' + y'^2 \right) \frac{\partial}{\partial x'} + \left(-2x'^2 - 2x'y' + 2y'^2 \right) \frac{\partial}{\partial y'}$$

We extend the definition of v to V thanks to the above expression:

```
Entrée [140]: v.add_comp_by_continuation(eV, W, chart=stereoS)
```

```
Entrée [141]: v.display(eV)
```

Out[141]:
$$v = \left(-x'^2 + 4x'y' + y'^2 \right) \frac{\partial}{\partial x'} + \left(-2x'^2 - 2x'y' + 2y'^2 \right) \frac{\partial}{\partial y'}$$

At this stage, the vector field v is defined on the whole manifold \mathbb{S}^2 ; it has expressions in each of the two frames eU and eV which cover \mathbb{S}^2 :

```
Entrée [142]: print(v)
              v.display(eU)
```

Vector field v on the 2-dimensional differentiable manifold S^2

Out[142]:
$$v = \frac{\partial}{\partial x} - 2 \frac{\partial}{\partial y}$$

Entrée [143]: `v.display(eV)`

Out[143]:
$$v = \left(-x'^2 + 4x'y' + y'^2\right) \frac{\partial}{\partial x'} + \left(-2x'^2 - 2x'y' + 2y'^2\right) \frac{\partial}{\partial y'}$$

According to the hairy ball theorem, v has to vanish somewhere. This occurs at the North pole:

Entrée [144]: `vN = v.at(N)`
`print(v)`

Vector field v on the 2-dimensional differentiable manifold S^2

Entrée [145]: `vN.display()`

Out[145]: $v = 0$

$v|_N$ is the zero vector of the tangent vector space $T_N S^2$:

Entrée [146]: `vN.parent()`

Out[146]: $T_N S^2$

Entrée [147]: `vN.parent() is S2.tangent_space(N)`

Out[147]: True

Entrée [148]: `vN == S2.tangent_space(N).zero()`

Out[148]: True

On the contrary, v is non-zero at the South pole:

Entrée [149]: `vS = v.at(S)`
`print(v)`

Vector field v on the 2-dimensional differentiable manifold S^2

Entrée [150]: `vS.display()`

Out[150]:
$$v = \frac{\partial}{\partial x} - 2 \frac{\partial}{\partial y}$$

Entrée [151]: `vS.parent()`

Out[151]: $T_S S^2$

Entrée [152]: `vS.parent() is S2.tangent_space(S)`

Out[152]: True


Entrée [153]: `vS != S2.tangent_space(S).zero()`

Out[153]: True

Let us plot the vector field v in terms of the stereographic chart $(U, (x, y))$, with the South pole S superposed:

```
Entrée [154]: v.plot(chart=stereoN, chart_domain=stereoN, max_range=4,
                    number_values=5, scale=0.5, aspect_ratio=1) + \
S.plot(stereoN, size=30, label_offset=0.2)
```

Out[154]:



The vector field appears homogeneous because its components w.r.t. the frame $\left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}\right)$ are constant:


```
Entrée [155]: v.display(stereoN.frame())
```

Out[155]:
$$v = \frac{\partial}{\partial x} - 2 \frac{\partial}{\partial y}$$

On the contrary, once drawn in terms of the stereographic chart $(V, (x', y'))$, v does no longer appears homogeneous:


```
Entrée [156]: v.plot(chart=stereoS, chart_domain=stereoS, max_range=4, scale=0.02,
                    aspect_ratio=1) + \
N.plot(chart=stereoS, size=30, label_offset=0.2)
```

Out[156]:



Finally, a 3D view of the vector field v is obtained via the embedding Φ :

```
Entrée [157]: graph_v = v.plot(chart=cart, mapping=Phi, chart_domain=spher,
                             number_values=11, scale=0.2)
show(graph_spher + graph_v, viewer=viewer3D)
```




Similarly, let us draw the first vector field of the stereographic frame from the North pole, namely $\frac{\partial}{\partial x}$:

```
Entrée [158]: ex = stereoN.frame()[1]
ex
```

```
Out[158]:  $\frac{\partial}{\partial x}$ 
```

```
Entrée [159]: graph_ex = ex.plot(chart=cart, mapping=Phi, chart_domain=spher,
                                number_values=11, scale=0.4, width=1,
                                label_axes=False)
show(graph_spher + graph_ex, viewer=viewer3D)
```




For the second vector field of the stereographic frame from the North pole, namely $\frac{\partial}{\partial y}$, we get

```
Entrée [160]: ey = stereoN.frame()[2]
ey
```


```
Out[160]:  $\frac{\partial}{\partial y}$ 
```

```
Entrée [161]: graph_ey = ey.plot(chart=cart, mapping=Phi, chart_domain=spher,
                                number_values=11, scale=0.4, width=1, color='red',
                                label_axes=False)
show(graph_spher + graph_ey, viewer=viewer3D)
```




We may superpose the two graphs, to get a 3D view of the vector frame associated with the stereographic coordinates from the North pole:

```
Entrée [162]: graph_frame = graph_spher + graph_ex + graph_ey + \
                N.plot(cart, mapping=Phi, label_offset=0.05, size=5) + \
                S.plot(cart, mapping=Phi, label_offset=0.05, size=5)
show(graph_frame + sphere(color='lightgrey', opacity=0.4), viewer=viewer3D)
```



The same scene rendered with Tachyon:

```
Entrée [163]: show(graph_frame + sphere(opacity=0.5), viewer='tachyon', figsize=10)
```



Vector fields acting on scalar fields

v and f are both fields defined on the whole sphere (respectively a vector field and a scalar field). By the very definition of a vector field, v acts on f :

```
Entrée [164]: vf = v(f)
              print(vf)
              vf.display()
```

Scalar field $v(f)$ on the 2-dimensional differentiable manifold S^2

$$\begin{aligned}
 \text{Out[164]: } v(f) : S^2 &\longrightarrow \mathbb{R} \\
 \text{on } U : (x, y) &\longmapsto -\frac{4(x-2y)}{x^4+2x^2y^2+y^4+1} \\
 \text{on } V : (x', y') &\longmapsto -\frac{4(x'^3-2x'^2y'+x'y'^2-2y'^3)}{x'^4+2x'^2y'^2+y'^4+1} \\
 \text{on } A : (\theta, \phi) &\longmapsto \frac{2((\cos(\phi)-2\sin(\phi))\cos(\theta)-\cos(\phi)+2\sin(\phi))\sin(\theta)}{\cos(\theta)^2+1}
 \end{aligned}$$

Values of $v(f)$ at the North pole, at the equator point E and at the South pole:

```
Entrée [165]: vf(N)
```

```
Out[165]: 0
```

Entrée [166]: `vf(E)`

Out[166]: 4

Entrée [167]: `vf(S)`

Out[167]: 0

1-forms

A 1-form on \mathbb{S}^2 is a field of linear forms on the tangent spaces. For instance it can be the differential of a scalar field:

Entrée [168]: `df = f.differential() ; print(df)`

1-form df on the 2-dimensional differentiable manifold \mathbb{S}^2

Entrée [169]: `df.display()`

Out[169]:
$$df = \left(-\frac{4x}{x^4 + 2x^2y^2 + y^4 + 1} \right) dx + \left(-\frac{4y}{x^4 + 2x^2y^2 + y^4 + 1} \right) dy$$

Entrée [170]: `print(df.parent())`

Module $\Omega^1(\mathbb{S}^2)$ of 1-forms on the 2-dimensional differentiable manifold \mathbb{S}^2

Entrée [171]: `df.parent()`

Out[171]: $\Omega^1(\mathbb{S}^2)$

The 1-form acting on a vector field:

Entrée [172]: `print(df(v)) ; df(v).display()`

Scalar field df(v) on the 2-dimensional differentiable manifold \mathbb{S}^2

Out[172]:
$$\begin{aligned} df(v) : \quad \mathbb{S}^2 &\longrightarrow \mathbb{R} \\ \text{on } U : \quad (x, y) &\longmapsto -\frac{4(x-2y)}{x^4+2x^2y^2+y^4+1} \\ \text{on } V : \quad (x', y') &\longmapsto -\frac{4(x'^3-2x'^2y'+x'y'^2-2y'^3)}{x'^4+2x'^2y'^2+y'^4+1} \\ \text{on } A : \quad (\theta, \phi) &\longmapsto -\frac{2((\cos(\phi)-2\sin(\phi))\cos(\theta)-3\cos(\phi)+6\sin(\phi))\sin(\theta)^3-4((\cos(\phi)-2\sin(\phi))\cos(\theta)-\cos(\phi)+2\sin(\theta)^4+2(\cos(\theta)-2)\sin(\theta)^2-4\cos(\theta)+4)}{\sin(\theta)^4+2(\cos(\theta)-2)\sin(\theta)^2-4\cos(\theta)+4} \end{aligned}$$

Let us check the identity $df(v) = v(f)$:

Entrée [173]: `df(v) == v(f)`

Out[173]: True

Similarly, we have $\mathcal{L}_v f = v(f)$:

Entrée [174]: `f.lie_derivative(v) == v(f)`

Out[174]: True

Curves in \mathbb{S}^2

In order to define curves in \mathbb{S}^2 , we first introduce the field of real numbers \mathbb{R} as a 1-dimensional smooth manifold with a canonical coordinate chart:

Entrée [175]: `R.<t> = RealLine() ; print(R)`

Real number line R

Entrée [176]: `R.category()`

Out[176]: **Smooth_R**

Entrée [177]: `dim(R)`

Out[177]: 1

Entrée [178]: `R.atlas()`

Out[178]: $[(\mathbf{R}, (t))]$

Let us define a **loxodrome of the sphere** in terms of its parametric equation with respect to the chart `spher = (A, (θ, φ))`

Entrée [179]: `c = S2.curve({spher: [2*atan(exp(-t/10)), t]}, (t, -oo, +oo), name='c')`

Curves in \mathbb{S}^2 are considered as morphisms from the manifold \mathbb{R} to the manifold \mathbb{S}^2 :

Entrée [180]: `c.parent()`

Out[180]: $\text{Hom}(\mathbf{R}, \mathbb{S}^2)$

Entrée [181]: `c.display()`

Out[181]: $c: \mathbf{R} \longrightarrow \mathbb{S}^2$

$$t \longmapsto (x, y) = \left(\cos(t) e^{\left(\frac{1}{10} t\right)}, e^{\left(\frac{1}{10} t\right)} \sin(t) \right)$$


$$t \longmapsto (x', y') = \left(\cos(t) e^{\left(-\frac{1}{10} t\right)}, e^{\left(-\frac{1}{10} t\right)} \sin(t) \right)$$

$$t \longmapsto (\theta, \phi) = \left(2 \arctan\left(e^{\left(-\frac{1}{10} t\right)}\right), t \right)$$

The curve c can be plotted in terms of stereographic coordinates (x, y) :

Entrée [182]: `c.plot(chart=stereoN, aspect_ratio=1)`


Out[182]:



We recover the well-known fact that the graph of a loxodrome in terms of stereographic coordinates is a **logarithmic spiral**.

Thanks to the embedding Φ , we may also plot c in terms of the Cartesian coordinates of \mathbb{R}^3 :

```
Entrée [183]: graph_c = c.plot(mapping=Phi, max_range=40, plot_points=200,
                             thickness=2, label_axes=False)
show(graph_spher + graph_c, viewer=viewer3D)
```



The **tangent vector field** (or **velocity vector**) to the curve c is

```
Entrée [184]: vc = c.tangent_vector_field()
vc
```

Out[184]: c'

c' is a vector field *along* \mathbb{R} taking its values in tangent spaces to \mathbb{S}^2 :

```
Entrée [185]: print(vc)
```

Vector field c' along the Real number line \mathbb{R} with values on the 2-dimensional differentiable manifold \mathbb{S}^2

The set of vector fields along \mathbb{R} taking their values on \mathbb{S}^2 via the differential mapping $c : \mathbb{R} \rightarrow \mathbb{S}^2$ is denoted by $\mathfrak{X}(\mathbb{R}, c)$; it is a module over the algebra $C^\infty(\mathbb{R})$:

```
Entrée [186]: vc.parent()
```

Out[186]: $\mathfrak{X}(\mathbb{R}, c)$

```
Entrée [187]: vc.parent().category()
```

Out[187]: $\text{Modules}_{C^\infty(\mathbb{R})}$

```
Entrée [188]: vc.parent().base_ring()
```

Out[188]: $C^\infty(\mathbb{R})$


A coordinate view of c' :

```
Entrée [189]: vc.display()
```

Out[189]:
$$c' = \left(\frac{1}{10} \cos(t) e^{(\frac{1}{10} t)} - e^{(\frac{1}{10} t)} \sin(t) \right) \frac{\partial}{\partial x} + \left(\cos(t) e^{(\frac{1}{10} t)} + \frac{1}{10} e^{(\frac{1}{10} t)} \sin(t) \right) \frac{\partial}{\partial y}$$


Let us plot the vector field c' in terms of the stereographic chart $(U, (x, y))$:

```
Entrée [190]: show(vc.plot(chart=stereoN, number_values=30, scale=0.5, color='red') +
                    c.plot(chart=stereoN), aspect_ratio=1)
```



A 3D view of c' is obtained via the embedding Φ :

```
Entrée [191]: graph_vc = vc.plot(chart=cart, mapping=Phi, ranges={t: (-20, 20)},
                                number_values=30, scale=0.5, color='red',
                                label_axes=False)
show(graph_spher + graph_c + graph_vc, viewer=viewer3D)
```



Riemannian metric on \mathbb{S}^2

The standard metric on \mathbb{S}^2 is that induced by the Euclidean metric of \mathbb{R}^3 . Let us start by defining the latter:

```
Entrée [192]: h = R3.metric('h')
h[1,1], h[2,2], h[3,3] = 1, 1, 1
h.display()
```

```
Out[192]: h = dX ⊗ dX + dY ⊗ dY + dZ ⊗ dZ
```

The metric g on \mathbb{S}^2 is the pullback of h associated with the embedding Φ :

```
Entrée [193]: g = S2.metric('g')
g.set( Phi.pullback(h) )
print(g)
```

Riemannian metric g on the 2-dimensional differentiable manifold S^2

Note that we could have defined g intrinsically, i.e. by providing its components in the two frames `stereoN` and `stereoS`, as we did for the metric h on \mathbb{R}^3 . Instead, we have chosen to get it as the pullback of h , as an example of pullback associated with some differential map.

The metric is a symmetric tensor field of type (0,2):

```
Entrée [194]: print(g.parent())
```

Module $T^{(0,2)}(S^2)$ of type-(0,2) tensors fields on the 2-dimensional differentiable manifold S^2

```
Entrée [195]: g.tensor_type()
```

```
Out[195]: (0, 2)
```

```
Entrée [196]: g.symmetries()
```

symmetry: (0, 1); no antisymmetry

The expression of the metric in terms of the default frame on S^2 (`stereoN`):

```
Entrée [197]: g.display()
```

```
Out[197]: g = \left( \frac{4}{x^4 + y^4 + 2(x^2 + 1)y^2 + 2x^2 + 1} \right) dx \otimes dx + \left( \frac{4}{x^4 + y^4 + 2(x^2 + 1)y^2 + 2x^2 + 1} \right) dy \otimes dy
```

We may factorize the metric components:

```
Entrée [198]: g[1,1].factor() ; g[2,2].factor()
```

```
Out[198]: \frac{4}{(x^2 + y^2 + 1)^2}
```

```
Entrée [199]: g.display()
```

```
Out[199]: g = \frac{4}{(x^2 + y^2 + 1)^2} dx \otimes dx + \frac{4}{(x^2 + y^2 + 1)^2} dy \otimes dy
```

A matrix view of the components of g in the manifold's default frame:

```
Entrée [200]: g[:]
```

```
Out[200]: \begin{pmatrix} \frac{4}{(x^2+y^2+1)^2} & 0 \\ 0 & \frac{4}{(x^2+y^2+1)^2} \end{pmatrix}
```

Display in terms of the vector frame $(V, (\partial_{x'}, \partial_{y'}))$:

```
Entrée [201]: g.display(stereoS.frame())
```

```
Out[201]: g = \left( \frac{4}{x'^4 + y'^4 + 2(x'^2 + 1)y'^2 + 2x'^2 + 1} \right) dx' \otimes dx' + \left( \frac{4}{x'^4 + y'^4 + 2(x'^2 + 1)y'^2 + 2x'^2 + 1} \right) dy' \otimes dy'
```

Entrée [202]: `g.display(spher.frame(), chart=spher)`

Out[202]: $g = d\theta \otimes d\theta + \sin(\theta)^2 d\phi \otimes d\phi$

The metric acts on vector field pairs, resulting in a scalar field:

Entrée [203]: `print(g(v,v))`

Scalar field $g(v,v)$ on the 2-dimensional differentiable manifold S^2

Entrée [204]: `g(v,v).parent()`

Out[204]: $C^\infty(\mathbb{S}^2)$

Entrée [205]: `g(v,v).display()`

Out[205]: $g(v,v) : \mathbb{S}^2 \longrightarrow \mathbb{R}$
on $U : (x,y) \longmapsto \frac{20}{x^4+y^4+2(x^2+1)y^2+2x^2+1}$
on $V : (x',y') \longmapsto \frac{20(x'^4+2x'^2y'^2+y'^4)}{x'^4+y'^4+2(x'^2+1)y'^2+2x'^2+1}$
on $A : (\theta,\phi) \longmapsto 5 \cos(\theta)^2 - 10 \cos(\theta) + 5$

The **Levi-Civita connection** associated with the metric g :

Entrée [206]: `nab = g.connection()
print(nab)
nab`

Levi-Civita connection nabla_g associated with the Riemannian metric g on the 2-dimensional differentiable manifold S^2

Out[206]: ∇_g

As a test, we verify that ∇_g acting on g results in zero:

Entrée [207]: `nab(g).display()`

Out[207]: $\nabla_g g = 0$

The nonzero Christoffel symbols of g (skipping those that can be deduced by symmetry on the last two indices) w.r.t. two charts:

Entrée [208]: `g.christoffel_symbols_display(chart=sphereN)`

Out[208]: $\Gamma^x_{xx} = -\frac{2x}{x^2+y^2+1}$
 $\Gamma^x_{xy} = -\frac{2y}{x^2+y^2+1}$
 $\Gamma^x_{yy} = \frac{2x}{x^2+y^2+1}$
 $\Gamma^y_{xx} = \frac{2y}{x^2+y^2+1}$
 $\Gamma^y_{xy} = -\frac{2x}{x^2+y^2+1}$
 $\Gamma^y_{yy} = -\frac{2y}{x^2+y^2+1}$

Entrée [209]: `g.christoffel_symbols_display(chart=spher)`

Out[209]: $\Gamma^\theta_{\phi\phi} = -\cos(\theta)\sin(\theta)$
 $\Gamma^\phi_{\theta\phi} = \frac{\cos(\theta)}{\sin(\theta)}$

∇_g acting on the vector field v :

Entrée [210]: `print(nab(v))`

Tensor field nabla_g(v) of type (1,1) on the 2-dimensional differentiable manifold S^2

Entrée [211]: `nab(v).display(stereoN.frame())`

Out[211]:
$$\nabla_g v = \left(-\frac{2(x-2y)}{x^2+y^2+1} \right) \frac{\partial}{\partial x} \otimes dx + \left(-\frac{2(2x+y)}{x^2+y^2+1} \right) \frac{\partial}{\partial x} \otimes dy + \left(\frac{2(2x+y)}{x^2+y^2+1} \right) \frac{\partial}{\partial y} \otimes dx + \left(-\frac{2}{x} \right) \frac{\partial}{\partial y} \otimes dy$$

Curvature

The Riemann tensor associated with the metric g :

Entrée [212]: `Riem = g.riemann()
print(Riem)
Riem.display()`

Tensor field Riem(g) of type (1,3) on the 2-dimensional differentiable manifold S^2

Out[212]:
$$\begin{aligned} \text{Riem}(g) = & \left(\frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \right) \frac{\partial}{\partial x} \otimes dy \otimes dx \otimes dy + \left(-\frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \right) \frac{\partial}{\partial x} \otimes dy \otimes dx \otimes dy \\ & + \left(-\frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \right) \frac{\partial}{\partial y} \otimes dx \otimes dx \otimes dy + \left(\frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \right) \frac{\partial}{\partial y} \otimes dx \otimes dx \otimes dy \end{aligned}$$

The components of the Riemann tensor in the default frame on \mathbb{S}^2 :

Entrée [213]: `Riem.display_comp()`

Out[213]:
$$\begin{aligned} \text{Riem}(g)^x_{yxy} &= \frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \\ \text{Riem}(g)^x_{yyx} &= -\frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \\ \text{Riem}(g)^y_{xxy} &= -\frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \\ \text{Riem}(g)^y_{yyx} &= \frac{4}{x^4+y^4+2(x^2+1)y^2+2x^2+1} \end{aligned}$$

The components in the frame associated with spherical coordinates:

Entrée [214]: `Riem.display_comp(spher.frame(), chart=spher)`

Out[214]:
$$\begin{aligned} \text{Riem}(g)^\theta_{\phi\theta\phi} &= \sin(\theta)^2 \\ \text{Riem}(g)^\theta_{\phi\phi\theta} &= -\sin(\theta)^2 \\ \text{Riem}(g)^\phi_{\theta\theta\phi} &= -1 \\ \text{Riem}(g)^\phi_{\theta\phi\theta} &= 1 \end{aligned}$$

Entrée [215]: `print(Riem.parent())`

Module $T^*(1,3)(S^2)$ of type-(1,3) tensors fields on the 2-dimensional differentiable manifold S^2

Entrée [216]: `Riem.symmetries()`

no symmetry; antisymmetry: (2, 3)

The Riemann tensor associated with the Euclidean metric h on \mathbb{R}^3 is identically zero:

Entrée [217]: `h.riemann().display()`

Out[217]: $\text{Riem}(h) = 0$

The Ricci tensor and the Ricci scalar:

Entrée [218]: `Ric = g.ricci()
Ric.display()`

Out[218]:
$$\text{Ric}(g) = \left(\frac{4}{x^4 + y^4 + 2(x^2 + 1)y^2 + 2x^2 + 1} \right) dx \otimes dx + \left(\frac{4}{x^4 + y^4 + 2(x^2 + 1)y^2 + 2x^2 + 1} \right) dy \otimes dy$$

Entrée [219]: `R = g.ricci_scalar()
R.display()`

Out[219]:
$$\begin{aligned} r(g) : \quad \mathbb{S}^2 &\longrightarrow \mathbb{R} \\ \text{on } U : \quad (x, y) &\longmapsto 2 \\ \text{on } V : \quad (x', y') &\longmapsto 2 \\ \text{on } A : \quad (\theta, \phi) &\longmapsto 2 \end{aligned}$$

Hence we recover the fact that (\mathbb{S}^2, g) is a Riemannian manifold of constant positive curvature.

In dimension 2, the Riemann curvature tensor is entirely determined by the Ricci scalar R according to

$$R^i_{jlk} = \frac{R}{2} (\delta^i_k g_{jl} - \delta^i_l g_{jk})$$

Let us check this formula here, under the form $R^i_{jlk} = -R g_{jlk} \delta^i_l$:

Entrée [220]: `delta = S2.tangent_identity_field()
Riem == - R*(g*delta).antisymmetrize(2,3)`

Out[220]: True

Similarly the relation $\text{Ric} = (R/2) g$ must hold:

Entrée [221]: `Ric == (R/2)*g`

Out[221]: True

The **Levi-Civita tensor** associated with g :

Entrée [222]: `eps = g.volume_form()
print(eps)
eps.display()`

2-form eps_g on the 2-dimensional differentiable manifold S^2

Out[222]:
$$\epsilon_g = \left(\frac{4}{x^4 + y^4 + 2(x^2 + 1)y^2 + 2x^2 + 1} \right) dx \wedge dy$$

Entrée [223]: `eps.display(spher.frame(), chart=spher)`

Out[223]: $\epsilon_g = \sin(\theta) d\theta \wedge d\phi$

The exterior derivative of the 2-form ϵ_g :

Entrée [224]: `print(eps.exterior_derivative())`

3-form deps_g on the 2-dimensional differentiable manifold S^2

Of course, since \mathbb{S}^2 has dimension 2, all 3-forms vanish identically:

Entrée [225]: `eps.exterior_derivative().display()`

Out[225]: $d\epsilon_g = 0$

Non-holonomic frames

Up to now, all the vector frames introduced on \mathbb{S}^2 have been coordinate frames. Let us introduce a non-coordinate frame on the open subset A . To ease the notations, we change first the default chart and default frame on A to the spherical coordinate ones:

Entrée [226]: `A.default_chart()`

Out[226]: $(A, (x, y))$

Entrée [227]: `A.default_frame()`

Out[227]: $\left(A, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}\right)\right)$

Entrée [228]: `A.set_default_chart(spher)`
`A.set_default_frame(spher.frame())`
`A.default_chart()`

Out[228]: $(A, (\theta, \phi))$

Entrée [229]: `A.default_frame()`

Out[229]: $\left(A, \left(\frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right)$

We define the new frame e by relating it to the coordinate frame $\left(\frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)$ via a field of tangent-space automorphisms:

Entrée [230]: `a = A.automorphism_field()`
`a[1,1], a[2,2] = 1, 1/sin(th)`
`a.display()`

Out[230]: $\frac{\partial}{\partial \theta} \otimes d\theta + \frac{1}{\sin(\theta)} \frac{\partial}{\partial \phi} \otimes d\phi$

Entrée [231]: `a[:]`

Out[231]: $\begin{pmatrix} 1 & 0 \\ 0 & \frac{1}{\sin(\theta)} \end{pmatrix}$

Entrée [232]: `e = spher.frame().new_frame(a, 'e')`
`print(e) ; e`

Vector frame (A, (e_1,e_2))

Out[232]: $(A, (e_1, e_2))$

Entrée [233]: `e[1].display()`

Out[233]: $e_1 = \frac{\partial}{\partial \theta}$

Entrée [234]: `e[2].display()`

Out[234]: $e_2 = \frac{1}{\sin(\theta)} \frac{\partial}{\partial \phi}$

Entrée [235]: `A.frames()`

Out[235]: $\left[\left(A, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y} \right) \right), \left(A, \left(\frac{\partial}{\partial x'}, \frac{\partial}{\partial y'} \right) \right), \left(A, \left(\frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right), \left(A, \left(\frac{\partial}{\partial X}, \frac{\partial}{\partial Y}, \frac{\partial}{\partial Z} \right) \right), (A, (e_1, e_2)) \right]$

The new frame is an orthonormal frame for the metric g :

Entrée [236]: `g(e[1],e[1]).expr()`

Out[236]: 1

Entrée [237]: `g(e[1],e[2]).expr()`

Out[237]: 0

Entrée [238]: `g(e[2],e[2]).expr()`

Out[238]: 1

Entrée [239]: `g[e,:]`

Out[239]: $\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$

Entrée [240]: `g.display(e)`

Out[240]: $g = e^1 \otimes e^1 + e^2 \otimes e^2$

Entrée [241]: `eps.display(e)`

Out[241]: $\epsilon_g = e^1 \wedge e^2$

It is non-holonomic: its structure coefficients are not identically zero:

Entrée [242]: `e.structure_coeff()[:]`

Out[242]: $\left[[[0, 0], [0, 0]], \left[\left[0, -\frac{\cos(\theta)}{\sin(\theta)} \right], \left[\frac{\cos(\theta)}{\sin(\theta)}, 0 \right] \right] \right]$

Entrée [243]: `e[2].lie_derivative(e[1]).display(e)`

Out[243]: $-\frac{\cos(\theta)}{\sin(\theta)} e_2$

while we have of course

Entrée [244]: `spher.frame().structure_coeff()[:]`

Out[244]: $[[[0, 0], [0, 0]], [[0, 0], [0, 0]]]$